

# Math 2500: Linear Algebra



Math 2500: Linear Algebra



# Contents

<b>Preface</b>	<b>v</b>
<b>1 Vectors, the Dot Product, and Matrices</b>	<b>1</b>
1.1 Vectors . . . . .	1
1.2 The Dot Product . . . . .	7
1.3 Matrices . . . . .	10
1.4 Getting Started with Sage . . . . .	16
1.5 Going Further with the Basic Objects . . . . .	17
<b>2 Linear Equations</b>	<b>19</b>
2.1 Three Geometric Models . . . . .	19
2.2 Solving Systems . . . . .	24
2.3 Elimination using Matrices . . . . .	28
2.4 Going Further with Elimination . . . . .	34
2.5 Matrix Algebra . . . . .	35
2.6 Matrix Inverses . . . . .	40
2.7 The $LU$ Decomposition . . . . .	45
2.8 Permutation Matrices . . . . .	50
2.9 Matrix Algebra: Going Further . . . . .	57
<b>3 Vector Spaces and Subspaces</b>	<b>59</b>
3.1 Subspaces of $\mathbb{R}^n$ . . . . .	59
3.2 The Nullspace . . . . .	63
3.3 Rank and the RREF . . . . .	67
3.4 Solving a System . . . . .	68
3.5 Going Further with Solving Systems . . . . .	69
3.6 Bases . . . . .	70
3.7 The Four Subspaces . . . . .	71
3.8 Going Further with the Fundamental Theorem . . . . .	72
<b>4 Orthogonality</b>	<b>73</b>
<b>5 Determinants</b>	<b>75</b>
<b>6 Eigendata and the Singular Value decomposition</b>	<b>77</b>



# Preface

This is a set of class notes designed to guide an inquiry-based course on linear algebra. This is not a complete resource! Rather, this text is meant to accompany the book *Introduction to Linear Algebra* by Gilbert Strang. Industrious students might also use it for a self-study course.

An important feature of this text is the integration of the Sage Mathematical Software System. Students in this course will learn to use Sage to perform long tedious computations (which linear algebra has in spades) and create visualizations. I thank the Sage community, in particular William Stein, for creating such a wonderful tool and making it open-source.

I also thank Rob Beezer for creating the MathBook project which makes this particular book project available on the web.





# Chapter 1

## Vectors, the Dot Product, and Matrices

This first chapter introduces the basic objects of linear algebra. You will meet vectors, the dot product, and matrices.

Vectors are a generalization of the concept of number. Where real numbers can help us model the geometry of points on a line, vectors will allow us to model the geometry of a plane, or (three-dimensional) space, or even "spaces" with higher dimensions. We begin by learning about the algebra of vectors, and making connections to the geometry.

The dot product is a funny kind of multiplication. It plays an important role in mathematics because it captures all of the basics of measurement. We shall learn how to use the dot product to measure lengths and angles. By its definition, the dot product is connected with the concept of a linear equation, so it will make frequent appearances in our work.

Matrices are another way to generalize the concept of number. (In fact, they generalize the concept of vector.) We start here by learning about the algebra of matrices. The whole rest of this course will focus on matrices, their uses, and their properties.

A running theme for this course is the use of the Sage mathematical software system. In order to get started, the fourth section of this course is dedicated to getting started with Sage using the SageMathCloud (SMC). You will make an account and run through an introductory workshop with SMC. Also, throughout this workbook you will find little embedded pieces of Sage code. These are implemented using the Sage SingleCell server. Most of these Sage cells are mutable. You can change the content in them and re-evaluate your new Sage code. I encourage you to play with these—it is a good way to learn the basics of Sage.

The fifth and final section of the chapter is a short assignment designed to consolidate learning. You will get a chance to practice your skills and to think more deeply about the concepts you have learned.

### 1.1 Vectors

#### The Assignment

- Read section 1.1 of *Strang* (pages 1-7).
- Read the following and complete the exercises below.

## Learning Objectives

Before class, a student should be able to:

- Add vectors.
- Multiply a vector by a scalar.
- Compute linear combinations.
- Draw pictures which correspond to the above operations.

At some point, a student should be able to:

- Solve linear combination equations involving unknown coefficients.
- Solve linear combination equations involving unknown vectors.

## Some Discussion

Algebraically, a **vector** is a stack of numbers in a set of parentheses or brackets, like this

$$\begin{pmatrix} 2 \\ 7 \\ 9 \end{pmatrix}, \text{ or } \begin{bmatrix} 2 \\ 7 \\ 9 \end{bmatrix}, \text{ or } (2 \ 7 \ 9).$$

The individual numbers are called the **components** or **entries** or **coordinates** of the vector. For example, 7 is the second component of the vectors above.

The first two vectors above are called **column** vectors because they are stacked vertically. The third is called a **row** vector because it is arranged horizontally. For this class, we will always use column vectors, but to save space, we might sometimes write them as row vectors. It is up to you to make the switch. (We will see later how this matters!)

Vectors can take lots of different sizes. The vectors above are all 3-vectors. Here is a 2-vector:

$$\begin{pmatrix} 71 \\ -12 \end{pmatrix}.$$

Here is a 4-vector:

$$\begin{pmatrix} \pi \\ 0 \\ -\pi \\ 1 \end{pmatrix}.$$

The main value in using vectors lies in their standard interpretations. Let's focus on 3-vectors for now. The vector  $\begin{pmatrix} a \\ b \\ c \end{pmatrix}$  can represent

- A point in space described in the standard three-dimensional rectangular coordinate system with  $x$  coordinate equal to  $a$ ,  $y$ -coordinate equal to  $b$  and  $z$  coordinate equal to  $c$ .
- An arrow in space which points from the origin  $(0,0,0)$  to the point  $(a,b,c)$ .
- An arrow in space which points from some point  $(x,y,z)$  to the point  $(x+a, y+b, z+c)$ .

## Operations on Vectors

There are two operations on vectors which are of utmost importance for linear algebra. (In fact, if your problem has these operations in it, there is a chance you are doing linear algebra already.)

**Scalar Multiplication** Given a number  $\lambda \in \mathbb{R}$  and a vector  $v = \begin{pmatrix} a \\ b \\ c \end{pmatrix}$ , we form the new vector

$$\lambda v = \begin{pmatrix} \lambda a \\ \lambda b \\ \lambda c \end{pmatrix}.$$

**Addition** Given a vector  $v = \begin{pmatrix} a \\ b \\ c \end{pmatrix}$  and a vector  $w = \begin{pmatrix} d \\ e \\ f \end{pmatrix}$  of the same size, we form their **sum**

$$v + w = \begin{pmatrix} a+d \\ b+e \\ c+f \end{pmatrix}$$

These operations have “obvious” generalizations to vectors of different sizes. Because things go entry-by-entry, these are often called **coordinate-wise** operations.

Combining these two operations gives us the notion of a *linear combination*. If  $\lambda$  and  $\mu$  are numbers and  $v$  and  $w$  are vectors of a common size, then the vector

$$\lambda v + \mu w$$

is a linear combination of  $v$  and  $w$ .

## Sage Instructions

**Basic Constructions** A vector in Sage is constructed by applying the `vector` command to a list. Lists are entered in square brackets with entries separated by commas, so the typical way to create a vector looks like this:

```
u = vector([1,1,2])
```

Notice that nothing was displayed. Sage just put the vector `u` into memory. We can ask for it by calling it.

```
u
```

```
(1, 1, 2)
```

Sage defaults to displaying vectors horizontally, which is different from how we normally write them by hand. This is okay. You will get used to it quickly.

Sage knows how to add, multiply by scalars, and form linear combinations, and the notation for it is just as easy as you would expect.

```
v = vector([-1,-1,2])
u + v
```

```
(0, 0, 4)
```

```
pi * u
```

```
(pi, pi, 2*pi)
```

```
3*u + 4*v
```

```
(-1, -1, 14)
```

If you ask Sage to plot a vector, you get this kind of picture:

```
plot(v)
```

And in two dimensions something similar...

```
a = vector([-1,1])  
plot(a)
```

If you find that you want a vector to have its tail someplace that is not the origin, use the `arrow` command.

```
plot(arrow([1,1],[2,3], color='red',
          arrowsize=2, width=2),
     figsize=5, aspect_ratio=1)
```

Note that Sage cut off some of this plot! Also, I used some options just to show them off. The `arrow` command works in three dimensions, too.

**Interactive Demonstrations** This is a Sage "interact." You can use this to explore the idea of linear combinations of 2-vectors.

```
@interact(layout= { 'top':[['a','c','e'],[['b','d','f'],[['l','m']]]})
def two_dim_plot(a=input_box(1,width=10),
    b=input_box(2,width=10),c=input_box(2,width=10),
    d=input_box(1,width=10),
        l=input_box(1,width=10), m=input_box(1,width=10),
            e=input_box(2,width=10),f=input_box(2,width=10)):
    two_dim = arrow([0,0], [a,b], color='red') +
        arrow([0,0],[c,d],color='blue')
    two_dim+= arrow([0,0], [l*a,l*b], color='red') +
        arrow([m*c,m*d], [l*a+m*c,l*b+m*d],color='red')
    two_dim+= arrow([0,0], [m*c,m*d], color='blue') +
        arrow([l*a,l*b], [l*a+m*c,l*b+m*d],color='blue')
    two_dim+= point([e,f],size=20,color='black',zorder=2)+
        arrow([l*a,l*b], [l*a+m*c,l*b+m*d],color='blue')
    two_dim+= text('vc=(a,b)', [e-.1,b+.1], color='red') +
        [c+.1,d-.1],color='purple')
    two_dim+= text('l*vl+m*w', [l*a+m*c+.1, l*b+m*d+.1],color='purple') +
        text('P=(e,f)', [e+.1,f+.1],color='black')
    two_dim+= arrow([0,0],[l*a+m*c,l*b+m*d],color='purple', arrowsize=1,
        width=1)
    two_dim.show(axes=True)
```

This is a different Sage interact. You can use this one to explore linear combinations of 2-vectors.

```
@interact(layout= { 'top':[['a','c','e'],[['b','d','f'],[['l','m']]]})
def two_dim_plot(a=input_box(1,width=10),
    b=input_box(2,width=10),c=input_box(2,width=10),
    d=input_box(1,width=10),
        l=input_box(1,width=10), m=input_box(1,width=10),
            e=input_box(2,width=10),f=input_box(2,width=10)):
    two_dim = arrow([0,0], [a,b], color='red') +
        arrow([0,0],[c,d],color='blue')
    two_dim+= arrow([0,0], [l*a,l*b], color='red') +
        arrow([m*c,m*d], [l*a+m*c,l*b+m*d],color='red')
    two_dim+= arrow([0,0], [m*c,m*d], color='blue') +
        arrow([l*a,l*b], [l*a+m*c,l*b+m*d],color='blue')
    two_dim+= point([e,f],size=20,color='black',zorder=2)+
        arrow([l*a,l*b], [l*a+m*c,l*b+m*d],color='blue')
    two_dim+= text('va=(a,b)', [e+1,b+1], color='red',
        [c+1,d-1],color='purple')
    two_dim+= text('l*va+m*w', [l*a+m*c+1, l*b+m*d+1],color='purple') +
        text('P=(e,f)', [e+1,f+1],color='black')
    two_dim+= arrow([0,0],[l*a+m*c,l*b+m*d],color='purple', arrowsize=1,
        width=1)
    two_dim.show(axes=True)
```

**Task 1.1.** Find an example of numbers  $\lambda$  and  $\mu$  so that

$$\lambda \begin{pmatrix} 1 \\ 2 \end{pmatrix} = \mu \begin{pmatrix} 2 \\ -1 \end{pmatrix}$$

or describe why no such example can exist.

**Task 1.2.** Find a vector  $b = \begin{pmatrix} b_1 \\ b_2 \end{pmatrix}$  so that

$$\begin{pmatrix} 2 \\ 7 \end{pmatrix} + \begin{pmatrix} b_1 \\ b_2 \end{pmatrix} = \begin{pmatrix} 10 \\ -3 \end{pmatrix}$$

or describe why no such example can exist.

**Task 1.3.** Find a vector  $b = \begin{pmatrix} b_1 \\ b_2 \end{pmatrix}$  so that this equation has at least one solution  $\lambda$

$$\begin{pmatrix} 1 \\ -2 \end{pmatrix} + \lambda \begin{pmatrix} b_1 \\ b_2 \end{pmatrix} = \begin{pmatrix} 2 \\ 3 \end{pmatrix}$$

or describe why no such example can exist.

**Task 1.4.** Give examples of numbers  $a$  and  $b$  such that

$$a \begin{pmatrix} 2 \\ 1 \end{pmatrix} + b \begin{pmatrix} 1 \\ 1 \end{pmatrix} = \begin{pmatrix} 7 \\ 5 \end{pmatrix}$$

or explain why no such numbers exist.

In the situations like the last exercise, the pair of numbers  $a, b$  is called a **solution** to the equation.

**Task 1.5.** Give an example of a vector  $X = \begin{pmatrix} x \\ y \end{pmatrix}$  so that the equation

$$a \begin{pmatrix} 2 \\ 1 \end{pmatrix} + bX = \begin{pmatrix} 7 \\ 5 \end{pmatrix}$$

has no solution  $(a, b)$ , or explain why no such example exists.

**Task 1.6.** Give an example of a number  $\lambda$  so that

$$\lambda \begin{pmatrix} 7 \\ -1 \\ 2 \end{pmatrix} + 3 \begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix} = \begin{pmatrix} 49 \\ -7 \\ 20 \end{pmatrix}$$

or explain why no such number exists.

**Task 1.7.** Give an example of numbers  $\lambda$  and  $\mu$  which are a solution to the equation

$$\lambda \begin{pmatrix} 7 \\ -1 \\ 2 \end{pmatrix} + \mu \begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix} = \begin{pmatrix} 49 \\ -7 \\ 20 \end{pmatrix}$$

or explain why no such solution exists.

**Task 1.8.** Give an example of a vector  $w = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$  so that the equation

$$a \begin{pmatrix} 1 \\ 1 \\ 0 \end{pmatrix} + b \begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix} = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$$

has no solution  $(a, b)$ , or explain why no such vector exists.

**Task 1.9.** Give an example of a vector  $w = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$  so that the equation

$$a \begin{pmatrix} 1 \\ 1 \\ 0 \end{pmatrix} + b \begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix} = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$$

has exactly one solution  $(a, b)$ , or explain why no such vector exists.

**Task 1.10.** Give an example of a vector  $X = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$  such that the equation

$$a \begin{pmatrix} 1 \\ 1 \\ 0 \end{pmatrix} + b \begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix} + c \begin{pmatrix} 1 \\ 1 \\ 1 \end{pmatrix} = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$$

has no solutions  $(a, b, c)$ , or explain why no such vector exists.

**Task 1.11.** Give an example of a vector  $X = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$  such that the equation

$$a \begin{pmatrix} 1 \\ 1 \\ 0 \end{pmatrix} + b \begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix} + c \begin{pmatrix} 1 \\ 1 \\ 1 \end{pmatrix} = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$$

has exactly one solution, or explain why no such vector exists.

**Task 1.12.** Give an example of a vector  $X = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$  such that the equation

$$a \begin{pmatrix} 1 \\ 0 \\ 2 \end{pmatrix} + b \begin{pmatrix} 0 \\ -1 \\ 0 \end{pmatrix} + c \begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} 3 \\ 7 \\ 7 \end{pmatrix}$$

has no solutions, or explain why no such vector exists.

## 1.2 The Dot Product

### The Assignment

- Read section 1.2 of *Strang*
- Read the following and complete the exercises below.

### Learning Objectives

Before class, a student should be able to

- Compute the dot product of two given vectors.
- Compute the length of a given vector.
- Normalize a given vector.
- Recognize that  $u \cdot v = 0$  is the same as " $u$  and  $v$  are orthogonal."
- Compute the angle between two given vectors using the cosine formula.

At some point, a student should be able to

- Interpret the statements  $u \cdot v < 0$  and  $u \cdot v > 0$  geometrically.
- Pass back and forth between linear equations and equations involving dot products.
- Make pictures of level sets of the dot product operation.

### Discussion

The dot product is a wonderful tool for encoding the geometry of Euclidean space, but it can be a bit mysterious at first. As Strang shows, it somehow holds all of the information you need to measure lengths and angles.

What does this weird thing have to do with linear algebra? A dot product with a "variable vector" is a way of writing a linear equation. For example,

$$\begin{pmatrix} 7 \\ 3 \\ -2 \end{pmatrix} \cdot \begin{pmatrix} x \\ y \\ z \end{pmatrix} = 7x + 3y - 2z.$$

Sometimes this will allow us to connect linear algebra to geometry, and use geometric thinking to answer algebraic questions.

### Sage and the dot product

#### Basic Commands

Sage has a built-in dot product command `u.dot_product(v)`. This will return the dot product of the vectors  $u$  and  $v$ .

```
u = vector([1,1,1]); v = vector([-1,0,1])
u.dot_product(v)
```

0

It also has a built-in command for computing lengths. Here sage uses the synonym "norm": `u.norm()`. Of course, you can also call this like a function instead of like a method: `norm(u)`.

```
u.norm(), v.norm()
```

```
(sqrt(3), sqrt(2))
```

There is no built-in command for angles. You just have to compute them using the cosine formula, like below. (I will break up the computation, but it is easy to do it all with one line.)

```
num = u.dot_product(v)
den = u.norm() * v.norm()
angle = arccos(num / den)
angle
```

```
1/2*pi
```

Of course, Sage's `arccos` command returns a result in *radians*. To switch to degrees, you must convert.

```
angle*180/pi
```

```
90
```

Often, it is helpful to normalize a vector. You can do this with the `normalized` method like this:

```
u.normalized()
```

```
(1/3*sqrt(3), 1/3*sqrt(3), 1/3*sqrt(3))
```

### Sage Interacts

Interacts would go here...

### Exercises about the Dot Product

**Task 1.13.** What shape is the set of solutions  $\begin{pmatrix} x \\ y \end{pmatrix}$  to the equation

$$\begin{pmatrix} 3 \\ 7 \end{pmatrix} \cdot \begin{pmatrix} x \\ y \end{pmatrix} = 5?$$

That is, if we look at all possible vectors  $\begin{pmatrix} x \\ y \end{pmatrix}$  which make the equation true, what shape does this make in the plane?

What happens if we change the vector  $\begin{pmatrix} 3 \\ 7 \end{pmatrix}$  to some other vector? What happens if we change the number 5 to some other number?

What happens if instead of 2-vectors, we use 3-vectors?

**Task 1.14.** Find an example of two 2-vectors  $v$  and  $w$  so that  $\begin{pmatrix} 1 \\ 2 \end{pmatrix} \cdot v = 0$  and  $\begin{pmatrix} 1 \\ 2 \end{pmatrix} \cdot w = 0$ , or explain why such an example is not possible.

**Task 1.15.** Let  $v = \begin{pmatrix} 3 \\ -1 \end{pmatrix}$ . Find an example of a pair of vectors  $u$  and  $w$  such that  $v \cdot u < 0$  and  $v \cdot w < 0$  and  $w \cdot u = 0$ , or explain why no such pair of vectors can exist.

**Task 1.16.** Find an example of three 2-vectors  $u$ ,  $v$ , and  $w$  so that  $u \cdot v < 0$  and  $u \cdot w < 0$  and  $v \cdot w < 0$ , or explain why no such example exists.

**Task 1.17.** Find an example of a number  $c$  so that

$$\begin{pmatrix} 1 \\ -1 \end{pmatrix} \cdot \begin{pmatrix} x \\ y \end{pmatrix} = c$$

has the vector  $\begin{pmatrix} 4 \\ 7 \end{pmatrix}$  as a solution, or explain why no such number exists.



**Task 1.18.** Let  $v = \begin{pmatrix} 2 \\ 1 \end{pmatrix}$  and  $w = \begin{pmatrix} -3 \\ 4 \end{pmatrix}$ . Find an example of a number  $c$  so that

$$v \cdot \begin{pmatrix} 1 \\ -1 \end{pmatrix} = c \quad \text{and} \quad w \cdot \begin{pmatrix} 1 \\ -1 \end{pmatrix} = c,$$

or explain why this is not possible.

**Task 1.19.** Let  $P = \begin{pmatrix} -3 \\ 4 \end{pmatrix}$ . Find an example of numbers  $c$  and  $d$  so that

$$\begin{pmatrix} 2 \\ -1 \end{pmatrix} \cdot P = c \quad \text{and} \quad \begin{pmatrix} 1 \\ -1 \end{pmatrix} \cdot P = d,$$

or explain why no such example is possible.

Now we move to three dimensions!

**Task 1.20.** Let  $V = \begin{pmatrix} 1 \\ 1 \end{pmatrix}$ . Find a unit vector of the form  $X = \begin{pmatrix} x \\ y \\ 0 \end{pmatrix}$  so that  $V \cdot X = \sqrt{2}$ , or explain why no such vector exists.

**Task 1.21.** Find an example of numbers  $c$ ,  $d$ , and  $e$  so that there is no solution vector  $X = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$  which simultaneously satisfies the three equations

$$\begin{pmatrix} 1 \\ 1 \\ 1 \end{pmatrix} \cdot X = c, \quad \begin{pmatrix} 2 \\ 2 \\ 2 \end{pmatrix} \cdot X = d, \quad \begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix} \cdot X = e,$$

or explain why no such numbers exist.

**Task 1.22.** Find an example of numbers  $c$ ,  $d$ , and  $e$  so that there is no solution vector  $X = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$  which simultaneously satisfies the three equations

$$\begin{pmatrix} 1 \\ 1 \\ 1 \end{pmatrix} \cdot X = c, \quad \begin{pmatrix} 0 \\ 1 \\ 1 \end{pmatrix} \cdot X = d, \quad \begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix} \cdot X = e,$$

or explain why no such numbers exist.

## 1.3 Matrices

### The Assignment

- Read *Strang* section 1.3 (pages 22-27).
- Read the following and complete the exercises below.

### Learning Goals

Before class starts, a student should be able to:

- multiply a matrix times a vector
  - as a linear combination of columns
  - as a set of dot products, row times column
- translate back and forth between our three representations
  - a system of linear equations,
  - a linear combination of vectors equation, and
  - a matrix equation  $Ax = b$ .
- Correctly identify the rows and columns of a matrix
- describe what is meant by a lower triangular matrix

At some point, as student should be comfortable with these concepts, which get a very brief informal introduction in this section:

- linear dependence and linear independence
- the inverse of a matrix

### Discussion

A **matrix** is a two-dimensional array of numbers like this:

$$A = \begin{pmatrix} 2 & 1 \\ 1 & 1 \end{pmatrix}.$$

Sometimes it helps to think of a matrix as a collection of its **rows** which are read across:

$$M = \begin{pmatrix} \longrightarrow \\ \longrightarrow \end{pmatrix}$$

and sometimes it helps to think of a matrix as a collection of its **columns** which are read down:

$$M = \begin{pmatrix} \downarrow & \downarrow \end{pmatrix}.$$

It is often more clear to describe a matrix by giving the sizes of its rows and columns. An  $m$  by  $n$  matrix is one having  $m$  rows and  $n$  columns. It is really easy to get these reversed, so be careful. For example, this is a  $2 \times 3$  matrix, because it has two rows and three columns:

$$B = \begin{pmatrix} 1 & 1 & 2 \\ 3 & 5 & 8 \end{pmatrix}$$

A matrix is called a **square** matrix when the number of rows and the number of columns is equal. The matrix  $A$  that I wrote down above is square because it is a  $2 \times 2$  matrix.

### Multiplying Matrices and Vectors

It is possible to multiply a matrix by a vector like this:

$$\begin{pmatrix} 1 & 1 & 2 \\ 3 & 5 & 8 \end{pmatrix} \begin{pmatrix} 13 \\ 21 \\ 34 \end{pmatrix} = \begin{pmatrix} 102 \\ 416 \end{pmatrix}$$

For this to work, it is absolutely crucial that the sizes match up properly. If the matrix is  $m$  by  $n$ , then the vector must have size  $n$ . In the above example  $m = 2$  and  $n = 3$ .

Later, we shall see that the word "multiplication" is not really the best choice here. It is better to think of the matrix as "acting on" the vector and turning it into a new vector. For now, the word multiplication will serve.

How exactly does one define this matrix-vector multiplication?

**Linear Combination of Columns Approach** The first way to perform the matrix-vector multiplication is to think of the vector as holding some coefficients for forming a linear combination of the columns of the matrix. In our example, it looks like this:

$$\begin{pmatrix} 1 & 1 & 2 \\ 3 & 5 & 8 \end{pmatrix} \begin{pmatrix} 13 \\ 21 \\ 34 \end{pmatrix} = 13 \begin{pmatrix} 1 \\ 3 \end{pmatrix} + 21 \begin{pmatrix} 1 \\ 5 \end{pmatrix} + 34 \begin{pmatrix} 2 \\ 8 \end{pmatrix} = \begin{pmatrix} 102 \\ 416 \end{pmatrix}$$

**Dot Products with the Rows Approach** The second way is to think of the matrix as a bundle of vectors lying along the rows of the matrix, and use the dot product. In our example above, this means that we consider the vectors

$$r_1 = \begin{pmatrix} 1 \\ 1 \\ 2 \end{pmatrix}, \quad r_2 = \begin{pmatrix} 3 \\ 5 \\ 8 \end{pmatrix}, \quad \text{and } v = \begin{pmatrix} 13 \\ 21 \\ 34 \end{pmatrix}$$

(notice I've rewritten the rows as columns) and then perform this kind of operation:

$$\begin{pmatrix} 1 & 1 & 2 \\ 3 & 5 & 8 \end{pmatrix} \begin{pmatrix} 13 \\ 21 \\ 34 \end{pmatrix} = \begin{pmatrix} r_1 \\ r_2 \end{pmatrix} v = \begin{pmatrix} r_1 \cdot v \\ r_2 \cdot v \end{pmatrix} = \begin{pmatrix} 102 \\ 416 \end{pmatrix}.$$

Two important remarks:

- Note that these operations only make sense if the sizes match up properly.
- Note that the two versions of the operation give you the same results.

### Matrix Equations

There are many situations in linear algebra that can be rewritten in the form of an equation that looks like this:

$$Av = b$$

where  $A$  is a matrix, and  $v$  and  $b$  are vectors. The interesting case is when we know  $A$  and  $b$ , but we want to find the unknown  $v$ . We will call this a **matrix-vector equation**.

Let's consider the case where you are given some **square** matrix  $A$ . Sometimes one can find another matrix  $B$  so that no matter what vector  $b$  is chosen in the matrix-vector equation above, the solution vector takes the form  $v = Bb$ . When

this happens, we say that  $A$  is **invertible** and call  $B$  the **inverse** of  $A$ . It is common to use the notation  $A^{-1}$  in place of  $B$ . This is a wonderful situation to be in! Eventually, we will want to figure out some test for when a given matrix is invertible, and find some ways to compute the inverse.

### A Note about Vectors

This reading also has a brief introduction to the idea of a set of vectors being **linearly dependent** or **linearly independent**. Strang is coy about the precise definition, so here it is:

A set of vectors  $v_1, v_2, \dots, v_n$  is called **linearly dependent** when there is some choice of numbers  $a_1, a_2, \dots, a_n$  which are not all zero so that the linear combination

$$a_1 v_1 + a_2 v_2 + \dots + a_n v_n = 0$$

A set of vectors which is not linearly dependent is called **linearly independent**.

This is a little funny the first time you read it. Note that for any set of vectors, you can make a linear combination of those vectors come out as 0. Simply choose all of the coefficients to be zero. But that is so easy to do we call it **trivial**. What the definition is asking is that we find a *nontrivial linear combination of the vectors to make zero*.

### Sage and Matrices

Sage has many useful commands for working with linear algebra, and given the central role played by matrices in this subject, there are lots of things Sage can do with matrices. We'll focus here on just basic construction and matrix-vector multiplication.

#### The matrix construction command

The command to construct a matrix is pretty straightforward. One types `matrix(r, c, [list of entries])` where  $r$  is the number of rows and  $c$  is the number of columns. The entries should be read across the rows starting with the top row.

```
A = matrix(2,3, [1,2,3,5,8,13]); A
```

```
[ 1  2  3]
[ 5  8 13]
```

If you wish, you can structure that list of entries to be a list of lists, where each sublist is a row in your matrix.

```
B = matrix(2,3, [[1,2,3], [5,8,13]]); B
```

```
[ 1  2  3]
[ 5  8 13]
```

Every once in a while, it might matter to you what kinds of numbers you put into the matrix. Sage will let you specify them by putting in an optional argument like this: `matrix(number type, r, c, [list of entries])`

```
C = matrix(ZZ, 2, 2, [2,1,1,1])
C # the best matrix
```

```
[2 1]
[1 1]
```

The notation `ZZ` means "integers." There are other sets of numbers here:

- **QQ** the rational numbers (with exact arithmetic)
- **RR** the real numbers (with computer precision arithmetic)
- **CC** the complex numbers
- **AA** the set of all algebraic numbers, that is, all of the numbers that are roots of some polynomial with integer coefficients

You can find out what kind of entries a matrix thinks it has by calling the `.parent()` method on it.

```
A.parent()
# this should say something about the integers
```

Full MatrixSpace of 2 by 3 dense matrices over Integer Ring

```
D = matrix(QQ, 3,3, [[1,0,1],[2/3, 1, 0],[0,0,9/5]])
# this should say something about the rationals
D.parent()
```

```
[ 1  0  1]
[2/3  1  0]
[ 0  0 9/5]
Full MatrixSpace of 3 by 3 dense matrices over Rational Field
```

### Building a matrix from rows or columns

It is possible to build a matrix by bundling together a bunch of vectors, too. Let's start with an example made using rows.

```
v1 = vector([2,1]); v2= vector([3,4])
# construct E with rows v1 and v2, then display
E = matrix([ v1, v2]); E
```

```
[2 1]
[3 4]
```

Sage prefers rows. I wish it were the other way, but I am sure there is a good reason it prefers rows. If you want to make a matrix whose columns are the vectors `v1` and `v2`, you can use the `transpose` method. We'll talk more about the operation of transpose later, but it basically "switches rows for columns and vice versa."

```
F = matrix([v1,v2]).transpose(); F
```

```
[2 3]
[1 4]
```

### Matrix action on vectors

Of course, Sage knows how to perform the action of a matrix on a vector.

```
C; v1
```

```
[2 1]
[1 1]
(2, 1)
```

```
C*v1
```

(5, 3)

And if you get the sizes wrong, it will return an error.

```
A; v1
```

```
[ 1  2  3]
[ 5  8 13]
(2, 1)
```

```
A*v1
```

```
Error in lines 1-1
```

```
...
```

```
TypeError: unsupported operand parent(s) for '*': 'Full_MatrixSpace_of_
2_by_3_dense_matrices_over_Integer_Ring' and 'Ambient_free_module_
of_rank_2_over_the_principal_ideal_domain_Integer_Ring'
```

If you really need it, Sage can tell you about inverses.

```
A.is_invertible()
```

```
False
```

```
C.is_invertible()
```

```
True
```

```
C.inverse()
```

```
[ 1 -1]
[-1  2]
```

## Exercises

**Task 1.23.** Make an example of a matrix  $\begin{pmatrix} 1 & \bullet \\ -1 & \bullet \end{pmatrix}$  so that the equation

$$\begin{pmatrix} 1 & \bullet \\ -1 & \bullet \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 1 \\ -1 \end{pmatrix}$$

has exactly one solution, or explain why this is not possible.

Interpret this as a statement about 2-vectors and draw the picture which corresponds.

**Task 1.24.** Make an example of a matrix  $\begin{pmatrix} 4 & 8 & \bullet \\ 3 & 6 & \bullet \\ 1 & 2 & \bullet \end{pmatrix}$  so that the equation

$$\begin{pmatrix} 4 & 8 & \bullet \\ 3 & 6 & \bullet \\ 1 & 2 & \bullet \end{pmatrix} \begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} 8 \\ 6 \\ 2 \end{pmatrix}$$

has exactly one solution, or explain why this is not possible.

Interpret this as a statement about 3-vectors and draw the picture which corresponds.

**Task 1.25.** Make an example of a matrix  $\begin{pmatrix} 2 & -1 \\ \bullet & \bullet \end{pmatrix}$  so that the equation

$$\begin{pmatrix} 2 & -1 \\ \bullet & \bullet \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 7 \\ 3 \end{pmatrix}$$

has exactly one solution, or explain why this is not possible.

Interpret this as a statement about a pair of lines in the plane and draw the picture which corresponds.

**Task 1.26.** Make an example of a matrix  $\begin{pmatrix} 1 & 0 & 1 \\ 1 & 1 & 3 \\ \bullet & \bullet & \bullet \end{pmatrix}$  so that the equation

$$\begin{pmatrix} 1 & 0 & 1 \\ 1 & 1 & 3 \\ \bullet & \bullet & \bullet \end{pmatrix} \begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} 1 \\ 1 \\ 1 \end{pmatrix}$$

has no solutions, or explain why this is not possible.

Interpret this as a statement about a planes in space and draw the picture which corresponds.

**Task 1.27.** Find a triple of numbers  $x$ ,  $y$ , and  $z$  so that the linear combination

$$x \begin{pmatrix} 1 \\ 2 \\ 3 \end{pmatrix} + y \begin{pmatrix} 4 \\ 5 \\ 6 \end{pmatrix} + z \begin{pmatrix} 7 \\ 8 \\ 9 \end{pmatrix}$$

yields the zero vector, or explain why this is not possible.

Rewrite the above as an equation which involves a matrix.

Plot the three vectors and describe the geometry of the situation.

**Task 1.28.** The vectors

$$r_1 = \begin{pmatrix} 1 \\ 4 \\ 7 \end{pmatrix}, \quad r_2 = \begin{pmatrix} 2 \\ 5 \\ 8 \end{pmatrix}, \quad \text{and} \quad r_3 = \begin{pmatrix} 3 \\ 6 \\ 9 \end{pmatrix}$$

are linearly dependent because they lie in a common plane (through the origin). Find a normal vector to this plane.

Since the vectors are linearly dependent, there must be (infinitely) many choices of scalars  $x$ ,  $y$ , and  $z$  so that  $xr_1 + yr_2 + zr_3 = 0$ . Find two sets of such numbers.

**Task 1.29.** Consider the equation

$$\begin{pmatrix} 2 & 1 \\ 1 & 1 \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} b_1 \\ b_2 \end{pmatrix}.$$

We are interested in being able to solve this for  $x$  and  $y$  for any given choice of the numbers  $b_1$  and  $b_2$ . Figure out a way to do this by writing  $x$  and  $y$  in terms of  $b_1$  and  $b_2$ .

Rewrite your solution in the form

$$\begin{pmatrix} x \\ y \end{pmatrix} = b_1 \begin{pmatrix} \bullet \\ \bullet \end{pmatrix} + b_2 \begin{pmatrix} \bullet \\ \bullet \end{pmatrix}.$$

How is this related to the inverse of the matrix  $A = \begin{pmatrix} 2 & 1 \\ 1 & 1 \end{pmatrix}$ ?

**Task 1.30.** Find an example of a number  $c$  and a vector  $\begin{pmatrix} b_1 \\ b_2 \end{pmatrix}$  so that the equation

$$\begin{pmatrix} 3 & 51 \\ c & 17 \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} b_1 \\ b_2 \end{pmatrix}$$

does not have a solution, or explain why no such example exists.

Explain your solution in terms of

- lines in the plane,
- 2-vectors and linear combinations, and
- invertibility of a matrix.

## 1.4 Getting Started with Sage

### The Assignment

It is time to get comfortable with the basics of using Sage. Begin by setting up an account with SageMathCloud.

- Be sure that you have login access to computers in student labs across campus. Before class, drop by a campus lab and make sure your credentials work properly and you can sign in to a machine. For our tutorial session we will use one of the labs on the first floor of Wright Hall, so one of those would make a good choice.
- Go to the course web site and follow the links to the Sage Intro workshop.
- Complete the first two steps of the workshop set up process:
  1. Start: where you make an account at SageMathCloud.
  2. Get: where you create your first "project" and populate it with files for a tutorial.

### Learning Objectives

At the end of this assignment, a student should have an account at SageMathCloud and should be able to log into the service without trouble. The student will also have a first project with some files.



## 1.5 Going Further with the Basic Objects

### The Assignment

- Go back through the exercises in this chapter. Complete any items you did not complete the first time through. Prepare any that we have not discussed in class so that you will be ready to present them.

### Discussion

Now we take a short break to revisit and consolidate the learning you have done so far. Revisit the reading and the exercises you have done in Chapter One: Basic Objects. The important feature of this work should be learning to think about your own thinking. This sort of **meta-cognition** characterizes expert learners. Eventually, you want to be able to monitor your work at all times and recognize when you understand deeply and when you do not. This will allow you to self-correct.

To help you get started with meta-cognition, I listed learning goals in each section. To go further, you need to explicitly go through the process of reviewing what you can do and what you cannot. Here are some prompts to help you get started with this process.

- Review the learning goals from each section. Can you do the things described? Can you do them sometimes, or have you mastered them so you can do them consistently?
- Look through all of the tasks and go deeper into them. Can you connect each exercise to one of our pictures? Try to build a mental model of how the exercise and its solution work.
- If your first solution to an exercise involve a “guess-and-check” approach, can you now complete the exercise in a *purposeful* and systematic manner?
- Make a list of concepts or exercises that are not clear to you. Phrase each item in your list as a question, and make each question as specific as possible. Talk with fellow students or your instructor until you can answer your own questions.



## Chapter 2

# Linear Equations

This chapter is dedicated to the first major problem of linear algebra: solving systems of linear equations. Restated as a set of questions, we will consider these.

- What is the set of solutions to a given system of linear equations?
- When does a given system have solution, and when does it not?
- If there is a solution, how many solutions are there?
- What ways do we have of describing the collection of solutions?
- Is there a computationally effective way to find those solutions?

Though we begin with the first question, we answer the last one first. As we explore the process of finding solutions, we will start to build the tools we need to finish answering the other questions in a later chapter. In this chapter, we aim to get as complete understanding as we can for at least a special case: **square systems**.

We will begin with a study of the two ways that vectors let us make pictures of systems of linear equations. Then we take up a basic process for finding solutions, where matrices will appear as a convenient notational device. But as we dig a little further, matrices will become interesting in their own way, so we will study those. But what we study will relate back to the fundamental issue of solving systems of linear equations.

This chapter has two short review sections in it. One just after we learn about elimination, and another at the end of the chapter.

## 2.1 Three Geometric Models

### The Assignment

- Read section 2.1 of *Strang* (pages 31-40).
- Read the following and complete the exercises below

### Learning Goals

Before class, a student should be able to:

- Translate back and forth between the three algebraic representations:
  - A system of linear equations.

- An equation involving a linear combination of vectors.
- A matrix equation.

- Can write down the  $n \times n$  **identity matrix**.

Sometime in the near future, a student should be able to:

- Given a system, interpret and plot the “row picture”.
- Given a system, interpret and plot the “column picture”.
- Use a matrix as a model of a **transformation**, including stating the **domain** and the **range**.

## Discussion

Now we have a little experience with vectors and related things, it is time to be aware of what we have done so we can use it as a foundation for future work. So far, we have talked about two geometric interpretations for a system of linear equations, the **row picture** and the **column picture**.

Does the following picture make sense to you?




Figure 2.1: The three geometric models of linear algebra

A deep understanding of linear algebra will involve a level of comfort with each of these three views of the subject in the diagram, and also the ability to pass back and forth between them.

### The Transformational View

We have seen that matrices can be made to "act upon" vectors by a kind of multiplication. In particular, if  $A$  is an  $m \times n$  matrix, then  $A$  can be multiplied (on the left) with a column vector of size  $n$ , and the result is a column vector of size  $m$ .

This makes  $A$  into a kind of **function**. (We will use the synonyms **mapping** or **transformation**, too.) For every vector  $v$  of size  $n$ , the matrix  $A$  allows us to compute a new vector  $T_A(v) = Av$  of size  $m$ . This is the basic example of what we will eventually call a **linear transformation**.

$$\mathbb{R}^m \xrightarrow{T_A} \mathbb{R}^n$$

$$v \longmapsto Av.$$

One of our long term goals is to find a way to think about the geometry of linear algebra from this viewpoint, too.

### Sage and Plotting for Linear Algebra

There are a few new Sage commands that might be useful here. We have already seen how to take linear equations and turn them into vectors and then turn the vector equation into a matrix equation. But Sage can help us move in the other direction, too.

The keys are commands to pull out the rows and columns from a given matrix. Let's start with a simple situation where the matrix equation is

$$\begin{pmatrix} 2 & 1 \\ 1 & 1 \end{pmatrix} \begin{pmatrix} x & y \end{pmatrix} = \begin{pmatrix} 3 & 4 \end{pmatrix}.$$

```
A = matrix(QQ, 2,2, [2,1,1,1]); A
```

```
[2 1]
[1 1]
```

```
x, y = var('x,y')
X = vector([x,y]); X
```

```
(x, y)
```

```
b = vector([3,4]); b
```

```
(3, 4)
```

To get Sage to pull out the columns, we can use the `.columns()` method. If we want just one column, we can use the `.column()` method, but then we have to remember to specify which column we want.

```
A.columns() # this will return a list
```

```
[(2, 1), (1, )]
```

```
A.column(1)
```

```
(1, 1)
```

*Big important note:* Sage always numbers lists starting with zero. so the first element of every list is the 0 entry, and the second element is the 1 entry.

Now it is possible to make Sage do things like this:

```
column_plot = plot(A.column(0), color='red')
column_plot+= plot(A.column(1), color='blue')
column_plot+= plot(b, color='purple')
show(column_plot, figsize=5, aspect_ratio=1)
```

This is an example of the a column picture.

One can also pull out the rows with corresponding row methods. And if you recall the way that matrix multiplication works if you think of rows, you can make a row picture.

```
A.rows()
```

```
[(2, 1), (1, 1)]
```

```
expr1 = A.row(0).dot_product(X) == b[0]
expr2 = A.row(1).dot_product(X) == b[1]

print expr1
print expr2
```

```
2*x + y == 3
x + y == 4
```

And now the picture:

```
row_plot = implicit_plot(expr1, [x,-5,5], [y,-1,9], color='blue')
row_plot+= implicit_plot(expr2, [x,-5,5], [y,-1,9], color='red')
show(row_plot, axes=True)
```

## Exercises

**Task 2.1.** Make an example of a system of linear equations which some students might find challenging to change into an equation involving a linear combination. Explain what the challenge is and how you can think clearly to overcome it.

**Task 2.2.** Make an example of a linear combination equation which some students might find challenging to change into a system of linear equations. Explain what the challenge is and how you can think clearly to overcome it.

**Task 2.3.** Consider the matrix equation

$$\begin{pmatrix} 1 & 2 & 4 \\ 2 & 0 & 1 \end{pmatrix} \begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} -1 \\ 3 \end{pmatrix}.$$

1. Draw a diagram representing the row picture of this matrix equation.
2. Draw a diagram representing the column picture of this matrix equation.

**Task 2.4.** Make an example of a system of linear equations so that the corresponding column picture is about linear combinations of four 2-vectors becoming the zero vector.

**Task 2.5.** Find a linear combination equation so that the corresponding system of linear equations corresponds to finding the intersection of three lines in the plane.

**Task 2.6.** Find an example of a vector  $b$  so that the equation

$$\begin{pmatrix} -1 & 2 \\ 5 & -9 \end{pmatrix} v = b$$

has no solution  $v$ , or explain why it is impossible to find such an example.

**Task 2.7.** *In each of the below, find an example of a matrix  $B$  which has the described effect.*

1.  $B \begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} y \\ x \end{pmatrix}$

2. *Rotates vectors through  $45^\circ$  counter-clockwise.*

3. *Reflects vectors across the  $y$ -axis.*

4.  $B \begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} x + y \\ y \end{pmatrix}.$

## 2.2 Solving Systems

### The Assignment

- Read section 2.2 of Strang (pages 45-51).
- Read the following and complete the exercises below.

### Learning Goals

Before class, a student should be able to do the following things.

- Clearly state and use the following vocabulary words: pivot, multiplier, triangular matrix, back substitution, singular, non-singular

Sometime after class, a student should be able to the following things.

- Perform elimination to put a system of linear equations into triangular form.
- Solve small systems by hand.
- Explain the two failure modes for elimination, and describe which leads to no solutions, and which leads to infinitely many solutions.
- Solve larger systems with the help of a computer algebra package (Sage).

### Discussion: Elimination for Solving Systems of Linear Equations

Now we begin the process of learning how to solve a system of linear equations systematically through a process called **elimination**.

#### Some terminology

A typical system looks something like this:

$$\begin{cases} 3x_1 + 2x_2 - \pi x_3 = 0 \\ -4x_1 - 33x_2 + x_3 = 12 \end{cases}$$

This situation is *two* equations in *three* unknowns. The unknowns here are the three numbers  $x_1$ ,  $x_2$  and  $x_3$  for which we search. Usually, we bundle the numbers together as a vector  $(x_1, x_2, x_3)$ . If we can find a vector which makes all of the equations true simultaneously, we call that vector a **solution**.

Keep in mind that the process involves eliminating instances of the variable below **pivots**. Strang describes the process pretty well, and gives good examples. What Strang describes in this section is sometimes called **the forward pass** elimination.

Watch out for situations which are **singular** in that they have fewer pivots than unknowns. A system is called **non-singular** if it has as many pivots as unknowns.

#### Keeping track of things

Playing with all of the equations is nice, but all that really matters is the collection of coefficients, and the numbers on the right hand sides of the equal signs. Experienced solvers get tired of copying notation from line to line in a computation, so they only keep track of the matrix of coefficients, **augmented** by the vector on the right-hand side. In the example above, that augmented matrix is

$$\left( \begin{array}{ccc|c} 3 & 2 & -\pi & 0 \\ -4 & -33 & 1 & 12 \end{array} \right)$$



All of the row operations can be performed on just this augmented matrix, without losing any of the essential information.

## Sage and Row Operations

The process of elimination for systems of equations involves performing operations on the equations. When translated to matrix form, it involves operations on the rows of the coefficient matrix. The corresponding matrix methods come in two types.

The first type of method modifies the matrix “in place”, which means that it *Changes the input matrix*.

- `A.rescale_row(r, num)` multiplies row `r` by the factor of `num`.
- `A.swap_rows(r1, r2)` switches the places of rows `r1` and `r2`.
- `A.add_multiple_of_row(target, useful, num)`. This adds `num` times row `useful` to row `target`.

Throughout, please remember that Sage uses 0-based indexing! So the rows are labeled 0, 1, 2, ...

```
A = matrix(QQ, 3,3, [0,2,4, 1,1,5, 6,2,5]); A
```

```
[0 2 4]
[1 1 5]
[6 2 5]
```

```
A.swap_rows(0,1); A
```

```
[1 1 5]
[0 2 4]
[6 2 5]
```

```
A.add_multiple_of_row(2,0,-6)
A # this should add -6 times row 0 to row 2
```

```
[ 1  1  5]
[ 0  2  4]
[ 0 -4 -25]
```

```
A.rescale_row(1,1/2); A
```

```
[ 1  1  5]
[ 0  1  2]
[ 0 -4 -25]
```

```
A.add_multiple_of_row(2,1,4)
A # this should add 4 times row 2 to row 2
```

```
[ 1  1  5]
[ 0  1  2]
[ 0  0 -17]
```

```
A.rescale_row(2,-1/17); A
```

```
[1 1 5]
[0 1 2]
[0 0 1]
```

This just did the whole process of **forward pass elimination**. (Well, we did a bit more than Strang would. He wouldn't rescale the rows.)

Sometimes you do not want to change the matrix **A**. If instead, you want to leave **A** alone, you can use these methods, which return a new object and do not change **A**.

- `A.with_rescaled_row(r, num)`
- `A.with_swapped_rows(r1, r2)`
- `A.with_added_multiple_of_row(t, u, num)`

Let's do the same operations as above, but without changing **A**. This will mean making a bunch of new matrices. In fact, let's also change the name of our matrix to **B**

```
B = matrix(QQ, 3,3, [0,2,4, 1,1,5, 6,2,5]); B
```

```
[0 2 4]
[1 1 5]
[6 2 5]
```

```
B1 = B.with_swapped_rows(0,1); B1
```

```
[1 1 5]
[0 2 4]
[6 2 5]
```

```
B2 = B1.with_added_multiple_of_row(2,0,-6)
B2 # this should add -6 times row 0 to row 2
```

```
[ 1  1  5]
[  0  2  4]
[  0 -4 -25]
```

```
B3 = B2.with_rescaled_row(1,1/2); B3
```

```
[ 1  1  5]
[  0  1  2]
[  0 -4 -25]
```

```
B4 = B3.with_added_multiple_of_row(2,1,4)
B4 # this should add 4 times row 2 to row 2
```

```
[ 1  1  5]
[  0  1  2]
[  0  0 -17]
```

```
B5 = B4.with_rescaled_row(2,-1/17); B5
```

```
[1 1 5]
[0 1 2]
[0 0 1]
```

This second option has some advantages. At any point, you can revise your work, because the original matrix is still in memory, and so are all of the intermediate steps. Let's display all six of the matrices at once to see that they all still exist.

```
B, B1, B2, B3, B4, B5
```

## Exercises

**Task 2.8.** Use the elimination method to transform this system into an easier one. (Can you make it triangular?) Circle the pivots in the final result.

$$\begin{cases} 2x + 3y + z = 8 \\ 4x + 7y + 5z = 20 \\ -2y + 2z = 0 \end{cases}$$

What two operations do you use to do this efficiently? Now use back substitution to solve the system.

**Task 2.9.** (Sage Exercise): Because the last system can be transformed in two operations, there are three equivalent systems generated through the process (the original, the intermediate, and the final).

Make row picture plots for each of the three systems. [Hint: Sage] How do the operations transform the pictures?

**Task 2.10.** Suppose that a system of three equations in three unknowns has two solutions  $(a, b, c)$  and  $(A, B, C)$ . Explain why the system must have other solutions than these two. Describe clearly two other solutions in terms of  $a, b, c, A, B, C$ .

**Task 2.11.** Find three examples of numbers  $a$  so that elimination will fail to give three pivots for this coefficient matrix:

$$A = \begin{pmatrix} a & 2 & 3 \\ a & a & 4 \\ a & a & a \end{pmatrix}$$

**Task 2.12.** How many ways can two lines in the plane meet? Make examples to represent as many qualitatively different situations as you can.

**Task 2.13.** Complete the following to make an example of a system of two equations in two unknowns which is singular but still has a solution, or explain why no such example exists.

$$\begin{cases} 2x + 3y = 1 \\ \bullet x + \bullet y = \bullet \end{cases}$$

**Task 2.14.** Complete the following to a system of three equations in three unknowns which is singular and does not have a solution, or explain why no such example exists.

$$\begin{cases} 3y - z = 1 \\ 2x - y + 3z = 0 \\ \bullet x + \bullet y + \bullet z = \bullet \end{cases}$$

**Task 2.15.** Complete the following to a system of three equations in three unknowns which is singular but still has a solution, or explain why no such example exists.

$$\begin{cases} x + y + z = 1 \\ 2x + y + 2z = 0 \\ \bullet x + \bullet y + \bullet z = \bullet \end{cases}$$

**Task 2.16.** How many ways can three planes in three dimensional space meet? Make examples to represent as many qualitatively different situations as you can.

## 2.3 Elimination using Matrices

### The Assignment

- Read section 2.3 of *Strang*.
- Read the discussion below and work out the exercises.

### Learning Goals

Before class, a student should be able to:

- Translate a system of linear equations into the form of an augmented matrix and back.
- Perform the forward pass elimination process to an augmented matrix.
- Multiply a pair of square matrices having the same size.
- Identify the matrix which performs the operation “add a multiple of row  $i$  to row  $j$ .”
- Identify the matrix which performs the operation “swap the places of row  $i$  and row  $j$ .”

Some time after class, a student should be able to:

- Use the steps from a forward pass elimination step to write a correct equation of the form

$$E_{\bullet} E_{\bullet} \cdots E_{\bullet} (A \mid b) = (U \mid b')$$

where  $U$  is an upper triangular matrix.

### Discussion: Elimination and Using Matrices as “Transformations”

Let us focus (for now) on square systems of equations, where the number of unknowns is equal to the number of equations.

#### The Four Ways to Write a System

Recall that there are three equivalent ways to write the typical linear algebra problem: (1) a system of linear equations to be solved simultaneously, (2) an equation expressing some linear combination of vectors with unknown coefficients as equal to another vector, and (3) a matrix equation.

Here is an example: This system of linear equations

$$\begin{cases} 3y + 2z = 8 \\ x - y + z = -1 \\ 3x + 2y + 3z = 1 \end{cases}$$

is equivalent to this equation using a linear combination of vectors

$$x \begin{pmatrix} 0 \\ 1 \\ 3 \end{pmatrix} + y \begin{pmatrix} 3 \\ -1 \\ 2 \end{pmatrix} + z \begin{pmatrix} 2 \\ 1 \\ 3 \end{pmatrix} = \begin{pmatrix} 8 \\ -1 \\ 1 \end{pmatrix}$$

and both of those are equivalent to this matrix equation

$$\begin{pmatrix} 0 & 3 & 2 \\ 1 & -1 & 1 \\ 3 & 2 & 3 \end{pmatrix} \begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} 8 \\ -1 \\ 1 \end{pmatrix}.$$

Each of these viewpoints has its advantages when talking about the geometry of linear algebra. But one way that things steadily improve as you move down the page is in the amount of notational baggage. From each version to the next, we lose repetitive bits of symbol that one can just remember from context. Often, this is taken one step further! We now throw away the unknowns, the equal signs and some of the parentheses surrounding the matrices and vectors, and just write an *augmented* matrix.

$$\left( \begin{array}{ccc|c} 0 & 3 & 2 & 8 \\ 1 & -1 & 1 & -1 \\ 3 & 2 & 3 & 1 \end{array} \right)$$

This is the minimum fuss way to keep track of all the information you need to solve the original system.

### Representing Elimination with Matrices

The process of elimination starts by performing operations on the system of equations. In the example above, one simplification we can make is to add  $-3$  times equation (ii) to equation (iii).

Then the new system looks like this:

$$\begin{cases} 3y + 2z = 8 \\ x - y + z = -1 \\ 5y = 4 \end{cases}$$

Let's translate that into the linear combination format:

$$x \begin{pmatrix} 0 \\ 1 \\ 0 \end{pmatrix} + y \begin{pmatrix} 3 \\ -1 \\ 5 \end{pmatrix} + z \begin{pmatrix} 2 \\ 1 \\ 0 \end{pmatrix} = \begin{pmatrix} 8 \\ -1 \\ 4 \end{pmatrix}.$$

What has happened to each of the vectors? Well, in each case, we have added  $-3$  times the second component to the third component. We have seen that one way to change vectors into other vectors is by (left-)multiplying them by matrices. Could we be so lucky that the operation "add  $-3$  times the second component to the third component" is representable by a matrix operation? YES. It is not hard to check that the matrix we need is

$$E = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & -3 & 1 \end{pmatrix}.$$

In fact, let's check it now for each of the four vectors we have in our system:

$$E \begin{pmatrix} 0 \\ 1 \\ 3 \end{pmatrix} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & -3 & 1 \end{pmatrix} \begin{pmatrix} 0 \\ 1 \\ 3 \end{pmatrix} = \begin{pmatrix} 0 \\ 1 \\ 0 \end{pmatrix}$$

and

$$E \begin{pmatrix} 3 \\ -1 \\ 2 \end{pmatrix} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & -3 & 1 \end{pmatrix} \begin{pmatrix} 3 \\ -1 \\ 2 \end{pmatrix} = \begin{pmatrix} 3 \\ -1 \\ 5 \end{pmatrix}$$

and

$$E \begin{pmatrix} 2 \\ 1 \\ 3 \end{pmatrix} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & -3 & 1 \end{pmatrix} \begin{pmatrix} 2 \\ 1 \\ 3 \end{pmatrix} = \begin{pmatrix} 2 \\ 1 \\ 0 \end{pmatrix}$$

and

$$E \begin{pmatrix} 8 \\ -1 \\ 1 \end{pmatrix} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & -3 & 1 \end{pmatrix} \begin{pmatrix} 8 \\ -1 \\ 1 \end{pmatrix} = \begin{pmatrix} 8 \\ -1 \\ 4 \end{pmatrix}.$$

Ha Ha! It all checks out. Those are the four vectors from our second system. This means that we can even use a simple substitution to rewrite things. (It is not obvious at the moment why this is helpful. Hang on a bit.)

$$x * E \begin{pmatrix} 0 \\ 1 \\ 3 \end{pmatrix} + y * E \begin{pmatrix} 3 \\ -1 \\ 2 \end{pmatrix} + z * E \begin{pmatrix} 2 \\ 1 \\ 3 \end{pmatrix} = E \begin{pmatrix} 8 \\ -1 \\ 1 \end{pmatrix}$$

Boy, it sure would be nice if we had a compact way to write that down...

### Matrix Multiplication

We make a compact way to write down that last equation by defining an operation of multiplying two matrices. If  $E$  and  $A$  are two matrices, we define their **matrix product** to be a new matrix as follows:

First, write  $A$  as a collection of columns  $v_i$

$$A = (v_1 \quad v_2 \quad v_3)$$

and then we declare that  $EA$  is the matrix made up of the columns  $Ev_i$  in the corresponding order.

$$EA = (Ev_1 \quad Ev_2 \quad Ev_3)$$

By way of example, we have already considered the matrices

$$E = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & -3 & 1 \end{pmatrix} \quad \text{and} \quad A = \begin{pmatrix} 0 & 3 & 2 \\ 1 & -1 & 1 \\ 3 & 2 & 3 \end{pmatrix}.$$

You should check that their product is now

$$EA = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & -3 & 1 \end{pmatrix} \begin{pmatrix} 0 & 3 & 2 \\ 1 & -1 & 1 \\ 3 & 2 & 3 \end{pmatrix} = \begin{pmatrix} 0 & 3 & 2 \\ 1 & -1 & 1 \\ 0 & 5 & 0 \end{pmatrix}$$

Finally, let's see how this influences our last two forms of the equations. The matrix form of our system was  $Ax = v$  where  $A$  is as above, and the vectors are  $v = \begin{pmatrix} 8 \\ -1 \\ 1 \end{pmatrix}$  and  $x = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$ . The neat part is that our new definition of matrix multiplication means that our elimination step transformed the equation

$$Ax = v \quad \text{or} \quad \begin{pmatrix} 0 & 3 & 2 \\ 1 & -1 & 1 \\ 3 & 2 & 3 \end{pmatrix} \begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} 8 \\ -1 \\ 1 \end{pmatrix}$$

into the newer equation

$$(EA)x = Ev \quad \text{or} \quad \begin{pmatrix} 0 & 3 & 2 \\ 1 & -1 & 1 \\ 0 & 5 & 0 \end{pmatrix} \begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} 8 \\ -1 \\ 4 \end{pmatrix}.$$

The same thing works for the augmented matrix. The augmented matrix form is really just writing down

$$(A \mid v) = \left( \begin{array}{ccc|c} 0 & 3 & 2 & 8 \\ 1 & -1 & 1 & -1 \\ 3 & 2 & 3 & 1 \end{array} \right)$$

and the elimination step changes this into

$$(EA \mid Ev) = \left( \begin{array}{ccc|c} 0 & 3 & 2 & 8 \\ 1 & -1 & 1 & -1 \\ 0 & 5 & 0 & 4 \end{array} \right).$$

### Matrices as Transformations

Take a moment and reflect on the key transition in what happens above. The most important thing that made it all work was that a matrix (the elimination matrix  $E$ ) was used to perform some sort of operation on vectors.

This is a key property of matrices. The matrix  $E$  defines a kind of function. For every vector  $w$  with three components, we can compute exactly one new vector  $EW$ , still with three components. This means that  $E$  defines a function from the set of 3-vectors to the set of 3-vectors.

### Sage and Matrix Multiplication

Sage has built-in matrix multiplication. You do the obvious thing and it works.

```
A = matrix(QQ, 2,2, [[2,1],[1,1]])
B = matrix(QQ, 2,2, [[0,3],[1,1]])
A, B
```

```
(
 [2 1]  [0 3]
 [1 1], [1 1]
)
```

```
A*B
```

You can check that it works with the way we defined matrix multiplication as a linear combination of vectors, too.

First, we define the column vectors by pulling out the entries from  $B$  and arranging them. To be sure, we ask Sage to display them as columns.

```
b1 = vector([B[0,0], B[1,0]])
b2 = vector([B[0,1], B[1,1]])
b1.column(), b2.column()
```

```
(
 [0]  [3]
 [1], [1]
)
```

```
A*b1.column(), A*b2.column()
```

```
(
 [1]  [7]
 [1], [4]
)
```

Now we can pile these rows into a matrix and then use the transpose to put them in columns.

```
C = matrix([A*b1, A*b2]).transpose()
C
```

```
[1 7]
 [1 4]
```

And we can double check everything by asking Sage if these things are equal.

```
C == A*B
```

```
True
```

This kind of test can be useful for checking our work! The discussion above has this multiplication:

```
D = matrix(QQ,3,4, [0,3,2,8,1,-1,1,-1,3,2,3,1]); D
```

```
[ 0  3  2  8]
[ 1 -1  1 -1]
[ 3  2  3  1]
```

```
E = matrix.identity(3)
E[2,1] = -3
E
```

```
[ 1  0  0]
[ 0  1  0]
[ 0 -3  1]
```

```
E*D
```

```
[ 0  3  2  8]
[ 1 -1  1 -1]
[ 0  5  0  4]
```

Ta-Da!!!

## Exercises

**Task 2.17.** In the main example above

$$\begin{cases} & 3y + 2z = 8 \\ x - y + z = -1 \\ 3x + 2y + 3z = 1 \end{cases}$$

we would rather have our first pivot in the upper left corner (i.e. the first row should have a non-zero coefficient for  $x$ ). This can be achieved by swapping the positions of rows (i) and (ii).

Find a matrix  $P_{12}$  so that multiplying by  $P_{12}$  on the left performs the corresponding row switching operation on the augmented matrix

$$\left( \begin{array}{ccc|c} 0 & 3 & 2 & 8 \\ 1 & -1 & 1 & -1 \\ 3 & 2 & 3 & 1 \end{array} \right)$$

**Task 2.18.** Consider the system

$$\begin{cases} 6x - y = 14 \\ 97x - 16y = 2/3 \end{cases}$$

Write this system in the other three forms: (1) an equation involving a linear combination of vectors; (2) an equation involving a  $2 \times 2$  matrix; (3) an augmented matrix.

**Task 2.19.** Perform an elimination step on the system from the last exercise to put the system in triangular form. You should get two pivots.

Write the new system in each of our four forms.

**Task 2.20.** Still working with the same system of equations, use Sage to make two column picture plots:

- One showing the three relevant column vectors from the original system.



- One showing the three relevant column vectors from the system after the elimination step.

You may find it helpful to look through the Sage examples in previous sections of this workbook.

**Task 2.21.** One more time, stay with the same system of equations. Use Sage to make two row picture plots:

- One showing the two relevant lines in the original system.
- One showing the two relevant lines from the system after the elimination step.

You may find it helpful to look through the Sage examples in previous sections of this workbook.

**Task 2.22.** Consider this system of three equations in three unknowns:

$$\begin{cases} -x + \frac{2}{3}y + z = 1 \\ x + 6y + z = 1 \\ 3x + 3z = 1 \end{cases}$$

Perform the elimination steps to transform this system into a triangular one.

Write down the corresponding matrices you use to perform each of these steps on the augmented matrix version of this system.

**Task 2.23.** Still working with the system of equations from the last task, use Sage to make two column picture plots:

- One showing the four relevant column vectors from the original system.
- One showing the four relevant column vectors from the system after the elimination step.

You may find it helpful to look through the Sage examples in previous sections of this workbook.

**Task 2.24.** One more time, stay with the system of equations from previous two tasks. Use Sage to make two row picture plots:

- One showing the two relevant lines in the original system.
- One showing the two relevant lines from the system after the elimination step.

You may find it helpful to look through the Sage examples in previous sections of this workbook.

## 2.4 Going Further with Elimination

### The Assignment

- Go back through the exercises in the first three sections of this chapter. Complete any items you did not complete the first time through. Prepare any that we have not discussed in class so that you will be ready to present them.

### Discussion

Now we take a short break to revisit and consolidate the learning you have done so far. Revisit the reading and the exercises you have done in Chapter Two: Linear Equations. The important feature of this work should be learning to think about your own thinking. This sort of **meta-cognition** characterizes expert learners. Eventually, you want to be able to monitor your work at all times and recognize when you understand deeply and when you do not. This will allow you to self-correct.

To help you get started with meta-cognition, I listed learning goals in each section. To go further, you need to explicitly go through the process of reviewing what you can do and what you cannot. Here are some prompts to help you get started with this process.

- Review the learning goals from each section. Can you do the things described? Can you do them sometimes, or have you mastered them so you can do them consistently?
- Look through all of the tasks and go deeper into them. Can you connect each exercise to one of our pictures? Try to build a mental model of how the exercise and its solution work.
- If your first solution to an exercise involve a “guess-and-check” approach, can you now complete the exercise in a *purposeful* and systematic manner?
- Make a list of concepts or exercises that are not clear to you. Phrase each item in your list as a question, and make each question as specific as possible. Talk with fellow students or your instructor until you can answer your own questions.

## 2.5 Matrix Algebra

### The Assignment

- Read section 2.4 of *Strang*.
- Read the following and complete the exercises below.

### Learning Goals

Before class, a student should be able to:

- Add and subtract matrices of the same size.
- Multiply matrices of appropriate sizes by one method.
- Compute powers  $A^p$  of a given square matrix  $A$ .
- Use the distributive law for matrix multiplication and matrix addition correctly.

Sometime after our meeting, a student should be able to:

- Multiply block matrices.
- Multiply matrices by *three* methods.
- Give examples to show how matrix multiplication is not like ordinary multiplication of real numbers: including the trouble with commutativity, and the difficulty with inverses.

### Discussion on Matrix Algebra

At the simplest level, this section is just about how to deal with the basic operations on matrices. We can add them and we can multiply them. We have already encountered matrix multiplication, and addition is even more natural.

But a subtle and important thing is happening here. Matrices are taking on a life of their own. They are becoming first class objects, whose properties are interesting and possibly useful.

This is an instance of the beginnings of *Modern Algebra*, which is the study of the algebraic structures of abstracted objects. In this case, we study whole collections of matrices of a common shape, and we try to treat them like generalized numbers. Then the natural questions are how much like “regular numbers” are these matrices?

Addition is about as well-behaved as you can expect, but multiplication is a bit trickier. Suddenly, two properties of multiplication for numbers don’t quite work for matrices:

- multiplication does not necessarily commute: It need not be the case that  $AB$  is the same as  $BA$ .
- we may not always have inverses: just because there is a matrix  $A$  which is not the zero matrix, it may not be the case that we can make sense of  $A^{-1}$  and get  $AA^{-1} = I$ .

## Sage and Matrix Algebra

Sage is aware of the basic matrix operations, and it won't let you get away with non-sense. Matrix multiplication and matrix addition are only defined if the dimensions of the matrices line up properly.

```
A = matrix(QQ, 2,3, [0,1,2,3,6,6])# A 2 by 3 matrix
B = matrix(QQ, 2,2, [4,2,3,1]) # 2 by 2 square matrix
C = matrix(QQ, 3,3, [2,1,2,1,2,1,2,1,2]) # a 3 by 3 square matrix
D = matrix(QQ, 2,3, [1,1,1,1,1,1]) # another 2 by 3 matrix
E = matrix(QQ, 3,2, [3,4,2,5,6,1]) # a 3 by 2 matrix
```

Let's see which of these Sage doesn't like. Can you predict, before evaluating the cells below, which of these will return an error?

```
A*B
```

```
B*A
```

```
[ 6 16 20]
[ 3  9 12]
```

```
A+B
```

```
A*C
```

```
[ 5  4  5]
[24 21 24]
```

```
C*A
```

```
A*D
```

```
A+D
```

## Sage and Matrix Addition

Matrix addition works a lot like addition of integers, as long as you fix a size first.

- There is a zero element.
- There are additive inverses (i.e. *negatives*).
- The operation is commutative.

```
#This constructs the zero matrix
Z = zero_matrix(2,3); Z
```

```
[0 0 0]
[0 0 0]
```

Let us add A and Z:

```
A + Z
```

```
[0 1 2]
[3 6 6]
```

We can check that adding Z doesn't change anything.

```
A + Z == A
```

```
True
```

And we can do the natural thing to get an additive inverse.

```
L = -A; L
```

```
[ 0 -1 -2]
[-3 -6 -6]
```

Finally, this last thing should return zero.

```
A + L
```

```
[0 0 0]
[0 0 0]
```

### Sage and Matrix Multiplication

Sage already has the structure of matrix multiplication built-in, and it can help with investigating the ways that matrix multiplication is different from regular multiplication of numbers.

We have seen above that Sage will not let us multiply matrices whose sizes do not match correctly. Of course, one way around that trouble is to stick to square matrices. But even there we can have trouble with the fact that matrix multiplication might not commute. It is rarely the case that  $XY = YX$ .

For those of you who will eventually study Modern Algebra, the collection of all  $n$ -square matrices is an example of a non-commutative ring with unit.

```
A*B, B*A
```

```
(
      [12 27 30]
[14  7] [15 32 34]
[57 48], [ 3 12 18]
)
```

Sage knows about the ring structure. We can check for an inverse.

```
B.is_invertible()
```

```
True
```

```
C.is_invertible()
```

```
False
```

And we can ask for the inverse in a couple of ways.

```
B.inverse()
```

```
[-1/2  1]
[ 3/2 -2]
```

```
B^(-1)
```

```
[-1/2  1]
[ 3/2 -2]
```



$$\begin{array}{c} \left[ \begin{array}{cc|cc} 0 & -1 & 0 & -1 & 2 \\ 1 & 0 & 1 & 0 & 3 \end{array} \right] \\ \hline \left[ \begin{array}{cc|cc} 2 & 3 & 2 & 3 & 1 \end{array} \right] \end{array}$$

## Exercises

**Task 2.25.** Make an example of a  $2 \times 3$  matrix and a  $3 \times 3$  matrix, and use this to demonstrate the three different ways to multiply matrices.

**Task 2.26.** Give an example of a pair of  $2 \times 2$  matrices  $A$  and  $B$  so that  $AB = 0$  but  $BA \neq 0$ , or explain why this is impossible.

**Task 2.27.** Give an example of a  $3 \times 3$  matrix  $A$  such that neither  $A$  nor  $A^2$  is the zero matrix, but  $A^3 = 0$ .

**Task 2.28.** Find all examples of matrices  $A$  which commute with both  $B = \begin{pmatrix} 1 & 0 \\ 0 & 0 \end{pmatrix}$  and  $C = \begin{pmatrix} 0 & 1 \\ 0 & 0 \end{pmatrix}$ . That is, find all matrices  $A$  so that  $AB = BA$  and  $AC = CA$ . How do you know you have all such matrices?

**Task 2.29.** Consider the matrix

$$A = \begin{pmatrix} 2 & 1 & 0 \\ -2 & 0 & 1 \\ 8 & 5 & 3 \end{pmatrix}.$$

Which elimination matrices  $E_{21}$  and  $E_{31}$  produce zeros in the  $(2, 1)$  and  $(3, 1)$  positions of  $E_{21}A$  and  $E_{31}A$ ?

Find a single matrix  $E$  which produces both zeros at once. Multiply  $EA$  to verify your result.

**Task 2.30.** Let's take a different view of the last computation. Block multiplication says that column 1 is eliminated by a step that looks like this one:

$$EA = \begin{pmatrix} 1 & 0 \\ -c/a & I \end{pmatrix} \begin{pmatrix} a & b \\ c & D \end{pmatrix} = \begin{pmatrix} a & b \\ 0 & D - cb/a \end{pmatrix}.$$

Here  $I$  is the  $2 \times 2$  identity matrix,  $D$  is a  $2 \times 2$  matrix, etc.

So, in the last exercise, what are  $a$ ,  $b$ ,  $c$  and  $D$  and what is  $D - cb/a$ ? Be sure to describe what shape each matrix has: the number of rows and columns.

**Task 2.31.** Suppose that we have already solved the equation  $Ax = b$  for the following three special choices of  $b$ :

$$Ax_1 = \begin{pmatrix} 1 \\ 0 \\ 0 \end{pmatrix}, \quad Ax_2 = \begin{pmatrix} 0 \\ 1 \\ 0 \end{pmatrix}, \quad \text{and} \quad Ax_3 = \begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix}.$$

If the three solutions are called  $x_1$ ,  $x_2$  and  $x_3$  and then bundled together to make the columns of a matrix

$$X = \begin{pmatrix} | & | & | \\ x_1 & x_2 & x_3 \\ | & | & | \end{pmatrix},$$

what is the matrix  $AX$ ? What does this mean about  $X$ ?

## 2.6 Matrix Inverses

### The Assignment

- Read section 2.5 of *Strang*.
- Read the following and complete the exercises below.

### Learning Goals

Before class, a student should be able to:

- State the definition of **invertible** matrix.
- Solve an equation  $Ax = b$  using the inverse of  $A$  if it exists.
- State how inverses and multiplication interact.
- Use Gauss-Jordan elimination to compute the inverse of a matrix.
- State a test for invertibility of square matrices using pivots.

Some time after class, a student should be able to:

- Describe the connection between Gauss-Jordan elimination and solving  $n$  different systems of equations.
- Describe the connection between Gauss-Jordan elimination, computing matrix inverses, and the process of elimination by matrix multiplication.
- State the definition of the determinant of a square matrix.
- State the connection between the determinant of a square matrix and invertibility.
- State the distinction between a matrix being **invertible** and a matrix being **singular**.

### Discussion

#### Matrix Inverses

The main point of this section is to start focusing on the first big problem in linear algebra. How can you tell, in advance, that a system of  $n$  equations in  $n$  unknowns will have a solution?

Of course, like all things we have been studying, this will have several different faces, all of which are equivalent. The one front and center right now is this: When does an  $n \times n$  square matrix have an inverse?

#### Finding an Inverse: Gauss-Jordan Elimination

There is an effective method for finding the inverse, and it is Gauss-Jordan elimination. (This is sometimes just called *Gaussian elimination*.) Essentially, you wish to solve  $n$  different systems  $Ax = b$  of size  $n \times n$  all at the same time, with specially chosen right hand sides.

The process is an algorithm, so it is very specific. If you do this some other way, you aren't doing Gauss-Jordan Elimination. The name is applied to the process.

*Gauss-Jordan Elimination*

- *Augment*: Tack on a copy of the identity matrix of the same size to the right hand side of your matrix. It should now look like  $(A \mid I)$ .



- *Forward Pass:* This is a nested procedure:
  - *preparation:* If necessary, use a row swap to make a non-zero entry in the upper left entry.
  - *make zeros:* The upper left entry is our first pivot. Use the operation of adding a multiple of the first row to the other rows to kill the entries below this first pivot.
  - *step down:* Step down to the second row and repeat the above, but ignoring rows and columns above and to the left. Repeat as necessary till you run out of rows.

If at any point in the process you get a row consisting of only zeros, perform a row switch to shuffle it to the bottom. When the forward pass is complete, you should have an upper triangular matrix.

- *Backward Pass:* This is also nested, like the forward pass, except that instead of working down and to the right, you begin at the lower right with the last pivot and work up and to the left. When complete, the matrix should have at most one non-zero entry in each row. This entry will be a pivot.
- *Rescale:* rescale rows to make the pivots into 1's.

At the end of the whole process, you should have something that looks like this:  $(I \mid B)$ . The wonderful part:  $B$  is the inverse of  $A$ . Well, almost. The process can fail! If along the line you find that the left hand block of your big augmented matrix doesn't have  $n$  pivots in it, then your matrix was not invertible.

What you have computed in the left hand block with the Gauss-Jordan elimination is the *reduced row-echelon form* of your original matrix.

### The Big Theorem: invertibility, singularity, and the determinant

What is the key?

**Theorem 2.2.** *An  $n \times n$  matrix  $A$  is invertible exactly when it has  $n$  pivots. Equivalently, its reduced row-echelon form has  $n$  non-zero entries down the diagonal. The inverse will be computed by Gauss-Jordan elimination.*

This is huge. The algorithm is not difficult, and it answers an important question exactly.

Note that we said a square matrix was *singular* when it did not have enough pivots. So what the above says is that a matrix is invertible if and only if it is non-singular.

### A simple test

We can use the above to make a simple numerical test of when a matrix is invertible. First do the forward pass of elimination to obtain an upper triangular matrix. Take the product of the diagonal entries. This will be zero if and only if one of the diagonal entries is zero, which will only happen if there are fewer than  $n$  pivots. This product is then helpful enough to test for invertibility, and so it deserves its own name: the **determinant**. We shall learn more about this quantity later.

### Sage and Gauss-Jordan Elimination

We have already seen that Sage has commands for constructing matrices and performing row operations. Those are the operations used to perform Gauss-Jordan

Elimination. But there are several interesting and useful commands in this neighborhood we have not yet discussed.

Let us construct my favorite matrix so we have something to play with.

```
A = matrix(QQ, 2,2, [2,1,1,1])
```

We can use the `.is_invertible()` method to check that `A` is invertible. In general, this method returns `True` or `False`.

```
A.is_invertible()
```

```
True
```

And we can get Sage to just compute the inverse for us.

```
A.inverse()
```

```
[ 1 -1]
[-1  2]
```

Just so we can see what happens if the matrix is not invertible, we try another matrix.

```
B = matrix(QQ, 2,2, [0,1,0,0])
B.is_invertible()
```

```
False
```

```
B.inverse()
```

```
Error in lines 1-1
...
ZeroDivisionError: input matrix must be nonsingular
```

We can also ask Sage to compute determinants with the `.determinant()` method.

```
A.determinant(), B.determinant()
```

```
(1,0)
```

Sage is also capable of computing the reduced row echelon form (the “rref”) of a matrix with the appropriately named `.rref()` method.

```
A.rref()
```

```
[1 0]
[0 1]
```

The method `.rref()` does not change the matrix `A`. There is another command which will work the same way for our purposes, `.echelon_form()`.

```
A.echelon_form()
```

```
[1 0]
[0 1]
```

There is a related command which *will find the rref and then update the matrix*. It is called `.echelonize()`. Because I don’t really want to mess with `A`, we will make a copy first.

```
C = copy(A) # fancy Python trick! (not so fancy)
C.echelonize()
```

Now we ask sage to print those out for us.

```
print A
print '\n'
print C
```

```
[2 1]
[1 1]

[1 0]
[0 1]
```

Now, we can be just a bit more hands-on with Gauss-Jordan elimination if we do it this way. We will combine commands we have used before to do this.

```
M = MatrixSpace(QQ, 2,2)
M(1) # this is the 2x2 identity
```

```
[1 0]
[0 1]
```

Now we do the algorithm.

```
D = A.augment(M(1))
D.rref()
```

```
[ 1  0  1 -1]
[ 0  1 -1  2]
```

That was good. But we only need the right-hand submatrix. We can get Sage to report just that!

```
E = D.rref().matrix_from_columns([2,3]); E
```

```
[ 1 -1]
[-1  2]
```

It is often convenient to chain methods together like this. Then you can read what happens from left to right.

## Exercises

Keep this in mind. The computations are simple, but tedious. Perhaps you want to use an appropriate tool.

**Task 2.32.** Use Gauss-Jordan elimination to find the inverse of the matrix  $A$  below.

$$A = \begin{pmatrix} 3 & 17 \\ 1 & 6 \end{pmatrix}$$

Be sure to clearly write down the operations you use and the matrices which perform the operations by left multiplication.

**Task 2.33.** Use Gauss-Jordan elimination to find the inverse of the matrix  $X$  below.

$$X = \begin{pmatrix} a & b \\ c & d \end{pmatrix}$$

Be sure to clearly write down the operations you use and the matrices which perform the operations by left multiplication.

**Task 2.34.** Use Gauss-Jordan elimination to find the inverse of the matrix  $B$  below.

$$B = \begin{pmatrix} 3 & 4 & -1 \\ 1 & 6 & 1 \\ 0 & 3 & -1 \end{pmatrix}$$

Be sure to clearly write down the operations you use and the matrices which perform the operations by left multiplication.

**Task 2.35.** Use Gauss-Jordan elimination to find the inverse of the matrix  $B$  below.

$$B = \begin{pmatrix} 0 & 3 & 4 & -1 \\ 0 & 1 & 6 & 1 \\ 2 & 0 & 3 & -1 \\ 5 & -1 & 1 & 3 \end{pmatrix}$$

Be sure to clearly write down the operations you use and the matrices which perform the operations by left multiplication.

**Task 2.36.** Use Gauss-Jordan elimination to find the inverse of the matrix  $D$  below.

$$D = \begin{pmatrix} 3 & 17 & -1 & 3 & 1 \\ 1 & 6 & -2 & 1 & 1 \\ 2 & 2 & 1 & -5 & 1 \\ 0 & 0 & 3 & 1 & -3 \\ -2 & 3 & 4 & 1 & 1 \end{pmatrix}$$

**Task 2.37.** Suppose that for the matrix  $D$  in the last exercise we imagine solving the matrix equation  $Dx = b$  for some vector  $b$  of the appropriate size. What might one mean by the row picture in this case? What might the column picture mean?

**Task 2.38.** Design a  $6 \times 6$  matrix which has the following properties:

- no entry equal to zero
- the reduced row echelon form should have exactly 5 pivots
- the 5 pivots should be different numbers
- no pair of rows should be scalar multiples of one another

Is your matrix invertible? How do you know? Does Sage say it is invertible?

## 2.7 The $LU$ Decomposition

### The Assignment

- Read section 2.6 of *Strang*.
- Read the following and complete the exercises below.

### Learning Goals

Before class, a student should be able to:

- Use Gaussian Elimination to find the  $LU$  and  $LDU$  decompositions of a matrix.
- Describe when the process of Gaussian Elimination will fail to produce an  $LU$  decomposition.

Sometime after class, a student should be able to:

- Solve a system of equations by using the  $LU$  decomposition and two triangular systems.
- Explain the connection between matrix elimination and the  $LU$  or  $LDU$  factorization of a matrix.

### Discussion: The $LU$ Decomposition of a Matrix

We now look at the ideas behind elimination from a more advanced perspective. If we think about the matrix multiplication form of the forward pass, we can realize it a *matrix decomposition theorem*:

**Theorem 2.3.** *Any square matrix  $A$  can be written as a product  $A = LU$  where  $L$  is a lower triangular matrix and  $U$  is an upper triangular matrix. Moreover, the matrix  $L$  will have 1's down its diagonal.*

There are three key observations that make this work:

- Each of the matrices  $E_{ij}$  that affects a row operation of the form *add a multiple of row  $i$  to row  $j$*  is an invertible matrix, with an easy to find inverse.
- If we make a sequence of row operations in the forward pass using matrices  $E_k$ , then we are essentially computing a big product

$$E_k \dots E_1 A = U$$

where each of the  $E_i$ 's is a lower triangular matrix and the matrix  $U$  is upper triangular. This can be rewritten as

$$A = (E_1^{-1} \dots E_k^{-1}) U.$$

Note that the inverses have to be done *in reverse order* for things to cancel out properly.

- Finally, the product  $L = E_1^{-1} \dots E_k^{-1}$  is really easy to compute, because its entries are simply the negatives of the multipliers we used to do the operations in the forward pass.

### A Nice Computational Result

One important output of this comes into play when we want to compute solutions to equations like  $Ax = b$ . Since we can write  $A = LU$ , then our equation can be split into two (big) steps:

1. First find the solution to the equation  $Ly = b$ .
2. Then find the solution to the equation  $Ux = y$ .

First, note that this is a good thing because both of the systems  $Ly = b$  and  $Ux = y$  are triangular. They can be solved by back substitution. In  $Ly = b$  you work from the top down, and in  $Ux = y$  you work from the bottom up.

Second, this works because following this process gives us a vector  $x$  which will satisfy this:

$$Ax = (LU)x = L(Ux) = Ly = b.$$

Third, this doesn't really save time when you only want to solve one equation  $Ax = b$ . But if you have lots of different values of  $b_i$ , and you want to solve all of the equations  $Ax = b_i$ , it becomes a lot faster to factor the matrix  $A = LU$  once and do two back substitutions for each value of  $b_i$ .

### Sage and The $LU$ Decomposition

A neat feature of linear algebra is that simple facts about solving equations have several different incarnations. This section contains the first big example: Gaussian Elimination leads to a multiplicative decomposition (a factorization) for matrices.

Each step of Gaussian elimination is a simple row operation, and if we do the process in the standard order, then the  $LU$  decomposition can be read out directly, without any extra computation.

First, let us recall how Sage can help us check bits of the three key observations above.

```
M = MatrixSpace(QQ,3,3)
One = M(1); One
```

```
[1 0 0]
[0 1 0]
[0 0 1]
```

Consider a matrix which performs an elementary row operation of the form “add a multiple of one row to another”. The matrix  $E$  below performs the operations *add -4 times row 2 to row 3*.

```
E = One.with_added_multiple_of_row(2,1,-4); E
```

```
[1 0 0]
[0 1 0]
[0 -4 1]
```

```
E.is_invertible()
```

```
True
```

```
E.inverse()
```

```
[1 0 0]
[0 1 0]
[0 4 1]
```

Note that the inverse just came from changing the sign of that one entry. This makes sense for the following reason: the opposite operation to “add  $-4$  times row 2 to row 3” should be “Add 4 times row 2 to row 3”. That is the way you undo the operation!

### Study Break: Try it yourself

Make your own  $3 \times 3$  matrix and check the whole procedure.

### Sage Commands to short-cut the process

Here is the basic command for getting Sage to compute the  $LU$  decomposition directly.

```
A = M([2,3,1,-1,3,5,6,5,4]); A
```

```
[ 2  3  1]
[-1  3  5]
[ 6  5  4]
```

```
A.LU()
```

```
(
[0 0 1] [ 1 0 0] [ 6 5 4]
[0 1 0] [-1/6 1 0] [ 0 23/6 17/3]
[1 0 0], [ 1/3 8/23 1], [ 0 0 -53/23]
)
```

Hold on, the output is three matrices. Not two, but three. One is upper triangular, one is lower triangular, but the first one is a **permutation matrix**. (It switches rows 1 and 3.) What is going on? If you perform a search in the Sage documentation, you find this page. There is a description of the command, and the first bit is something about a “pivoting strategy” and row swaps. But we don’t want row swaps.

By reading carefully, we can see what the way through is, too. We can specify our pivoting strategy by adding the keyword argument `pivot="nonzero"` inside the parentheses. Then the algorithm used will match the one Strang describes.

(If you are using SMC, you can access the help using many other ways. But a Google search for **Sage math "topic"** will hit the documentation pretty reliably.)

```
A.LU(pivot='nonzero')
```

```
(
[1 0 0] [ 1 0 0] [ 2 3 1]
[0 1 0] [-1/2 1 0] [ 0 9/2 11/2]
[0 0 1], [ 3 -8/9 1], [ 0 0 53/9]
)
```

Aaah! There we go, now the permutation part is the identity. Note that the command returns a “tuple”. This is a collection of things, kind of like a list. (Technical Python details omitted here.) To grab the information out, we assign the parts of that output to different names so we can use them.

```
P, L, U = A.LU(pivot='nonzero')
```

```
L # this is the lower triangular part
```

```
[ 1 0 0]
[-1/2 1 0]
[ 3 -8/9 1]
```

```
U # this is the upper triangular part
```

```
[ 2  3  1]
[ 0 9/2 11/2]
[ 0  0 53/9]
```

Those parts should be factors of  $A$ . We can check:

```
L*U # this should multiply to A
```

```
[ 2  3  1]
[-1  3  5]
[ 6  5  4]
```

And we can have Sage check if they are really equal.

```
L*U == A
```

```
True
```

### What about the $LDU$ decomposition?

For now, Sage has no built-in  $LDU$  decomposition.

### An insurmountable obstacle

Some matrices *require* permutations of rows. In these cases, we have to have some pivoting strategy *must* be employed. Consider this example.

```
B = M([0,2,2,1,3,1,1,1,1]); B
```

```
[ 0  2  2]
[ 1  3 -1]
[ 1  1  1]
```

```
B.LU(pivot='nonzero')
```

```
(
[0 1 0] [ 1  0  0] [ 1  3 -1]
[1 0 0] [ 0  1  0] [ 0  2  2]
[0 0 1], [ 1 -1  1], [ 0  0  4]
)
```

This has a row-swap permutation matrix, and *it must*. Since the (1,1) entry of  $B$  is zero, but numbers below that are not zero, we cannot use zero as a pivot. We'll sort out how to handle this in the next section.

### Exercises

Keep this in mind. The computations are simple, but tedious. Perhaps you want to use an appropriate tool.

**Task 2.39.** Consider the following system of 3 linear equations in 3 unknowns.

$$\begin{cases} x + y + z = 5 \\ x + 2y + 3z = 7 \\ x + 3y + 6z = 11 \end{cases}$$

Perform the forward pass of elimination to find an equivalent upper triangular system. Write down this upper triangular system. What three row operations do you need to perform to make this work?



Use the information you just found to write a matrix decomposition  $A = LU$  for the coefficient matrix  $A$  for this system of equations. (Be sure to multiply the matrices  $L$  and  $U$  to check your work.)

**Task 2.40.** Solve the two systems  $Ly = b$  and  $Ux = y$  obtained in the last exercise.

Solve the system  $Ax = b$  directly using Gauss-Jordan elimination (hint: use Sage) and make sure that the results are the same.

**Task 2.41.** Consider the matrix  $A$  below. Find the matrix  $E$  which transforms  $A$  into an upper triangular matrix  $EA = U$ . Find  $L = E^{-1}$ . Use this to write down the LU decomposition  $A = LU$  of  $A$ .

$$A = \begin{pmatrix} 2 & 1 & 0 \\ 0 & 4 & 2 \\ 6 & 3 & 5 \end{pmatrix}$$

**Task 2.42.** The matrix below is **symmetric**, because if you flip it across its main diagonal you get the same thing. Find the LDU triple decomposition of this symmetric matrix.

$$B = \begin{pmatrix} 2 & 4 \\ 4 & 11 \end{pmatrix}$$

**Task 2.43.** The matrix below is **symmetric**, because if you flip it across its main diagonal you get the same thing. Find the LDU triple decomposition of this symmetric matrix.

$$C = \begin{pmatrix} 1 & 4 & 0 \\ 4 & 12 & 4 \\ 0 & 4 & 0 \end{pmatrix}$$

**Task 2.44.** The matrix below is **symmetric**, because if you flip it across its main diagonal you get the same thing. Find the LU decomposition of this symmetric matrix.

$$D = \begin{pmatrix} a & a & a & a \\ a & b & b & b \\ a & b & c & c \\ a & b & c & d \end{pmatrix}$$

What conditions on the variables  $a$ ,  $b$ ,  $c$ , and  $d$  will guarantee that this matrix has four pivots?

**Task 2.45.** Find an example of a  $3 \times 3$  matrix  $A$  which has all of its entries non-zero, so that the LU decomposition has  $U = I$ , where  $I$  is the identity matrix, or explain why no such example exists.

## 2.8 Permutation Matrices

### The Assignment

- Read section 2.7 of *Strang*
- Read the following and complete the exercises below.

### Learning Goals

Before class, a student should be able to:

- Compute the transpose of a matrix.
- Correctly perform calculations where the transpose interacts with the operations of matrix sum, matrix product, and matrix inverse.
- Compute inner and outer products using the transpose.
- Decide if a matrix is symmetric or not.
- Recognize permutation matrices, and design permutation matrices which correspond to given row swaps.

Some time after class, a student should be able to:

- Find the  $LDL^T$  decomposition for symmetric matrices.
- Explain how the necessity of permuting rows during Gaussian elimination leads to the decomposition  $PA = LU$ .
- Explain why  $P^T = P^{-1}$  for permutation matrices.

### Discussion: Transposes, Symmetric Matrices, and Permutations

An important operation on matrices we have yet to encounter is called the **transpose**. If  $A$  is an  $m \times n$  matrix, the transpose  $A^T$  of  $A$  is made by changing the roles of the rows and the columns. The result  $A^T$  will be an  $n \times m$  matrix, because of this switch.

For now, the transpose will feel like some random thing, but its primary importance comes from its connection with the dot product. If we think of column vectors  $u$  and  $v$  of size  $n$  as if they are  $n \times 1$  matrices, then the dot product  $u \cdot v$  can be computed with a nice combination of matrix multiplication and the transpose:

$$u \cdot v = u^T v.$$

On the right, this is matrix multiplication! That makes sense because  $u^T$  is  $1 \times n$  and  $v$  is  $n \times 1$ . This means that the result is a  $1 \times 1$  matrix, i.e. a number.

Since the dot product contains all of the geometry of Euclidean space in it, the transpose becomes an important operation. I know that sounds weird, but the dot product contains all of the information we need to measure lengths and angles, so basically all of the *metric* information in Euclidean geometry is there.

### Algebraic results about the transpose

There are some key results about the way the transpose interacts with other matrix operations, each of these can be checked with some tedious computation:

- If  $A$  and  $B$  are matrices of the same shape, then  $(A + B)^T = A^T + B^T$ .
- If  $A$  and  $B$  are of sizes so that  $AB$  is defined, then  $(AB)^T = B^T A^T$ .
- If  $A$  is an invertible matrix, with inverse  $A^{-1}$ , then  $A^T$  is also invertible and it has inverse  $(A^T)^{-1} = (A^{-1})^T$ .

### Symmetric Matrices

A matrix  $A$  is called *symmetric* when  $A^T = A$ . These pop up in lots of interesting places in linear algebra. A neat result is that a symmetric matrix has a symmetric looking  $LDU$  decomposition:

$$\text{if } A^T = A, \text{ then } A = LDL^T.$$

That is, in the  $LDU$  decomposition,  $U = L^T$ .

There are several ways to get symmetric matrices. For example, if  $A$  is any matrix, the new matrix  $B = A^T A$  will be symmetric. (Check this.) Also, the matrix  $S = A^T + A$  will be symmetric.

### Permutation Matrices and Pivoting strategies in Gauss-Jordan Elimination

It is sometimes the case that Gauss-Jordan elimination requires a row swap. As we have seen, the operation of swapping a row can be achieved by left multiplying by a matrix of a special type. If we take a bunch of those and multiply them together, we still get a matrix which is in a special class: *the permutation matrices*.

A permutation matrix is square matrix having a single 1 in each column and in each row. A helpful property of permutation matrices is that they are invertible, and their inverses are the same as their transposes:

$$P^{-1} = P^T.$$

Gauss-Jordan elimination is easy enough to understand, now. It is time to let go of performing all those arithmetic operations by hand. So, permutation matrices become important for a different reason! Even if Gauss-Jordan elimination can be done without a row swap, it may be numerically better for a computer to swap out for a larger number as a pivot, so a row swap is used anyway. This partial pivoting strategy is encapsulated in most computer algebra algorithms in some way, and is part of the computation involved in computing a PLU decomposition. Strang has a decent discussion of the choices, below we will discuss how Sage handles this.

### Sage and Transposes, Symmetry, Permutations, and Pivots

There is a lot going on in this little section. At first glance, it is a bit intimidating. But we have seen most of the ideas before.

#### The Transpose

The transpose of a matrix is what you get by switching the roles of rows and columns. Sage has a simple method for this.

```
M = MatrixSpace(QQ, 3,3)
A = M([1,2,3,4,5,6,7,8,9]); A
```

```
[1 2 3]
[4 5 6]
[7 8 9]
```

```
A.transpose()
```

```
[1 4 7]
[2 5 8]
[3 6 9]
```

One place that the transpose is useful is in describing the dot product. Check this out.

```
u = vector([1,2,3])
v = vector([4,5,6])
u.dot_product(v)
```

```
32
```

```
U = u.column(); U # this puts u into a column matrix
```

```
[1]
[2]
[3]
```

To be sure, we check what the “parent” of  $U$  is.

```
U.parent()
```

```
Full MatrixSpace of 3 by 1 dense matrices over Integer Ring
```

See! Sage thinks of  $U$  as a matrix with 3 rows and 1 column.  
Now we do the same with  $v$

```
V = v.column()
V
```

```
[4]
[5]
[6]
```

Now the magic.

```
U.transpose()*V
```

```
32
```

```
V.transpose()*U
```

```
32
```

That is the dot product, but stuffed into a  $1 \times 1$  matrix!

### Other Properties

The transpose has other useful properties. Strang lists the big ones, including how the transpose interacts with matrix multiplication and matrix inverses.

## Symmetry

A matrix is called **symmetric** when it is equal to its transpose. Sage has some built-in commands for this.

```
B = M([2,1,0,1,1,0,0,0,1])
B
```

```
[2 1 0]
[1 1 0]
[0 0 1]
```

```
B.transpose()
```

```
[2 1 0]
[1 1 0]
[0 0 1]
```

```
B.is_symmetric()
```

```
True
```

```
C = M([1,0,1,1,1,1,0,0,0]); C
```

```
[1 0 1]
[1 1 1]
[0 0 0]
```

```
C.is_symmetric()
```

```
False
```

Strang notes a really neat property of symmetric matrices. Their *LDU* decompositions are nicer than average.

```
P, L, U = B.LU(pivot='nonzero')
```

```
P # here, things are good and no row swaps are needed
```

```
[1 0 0]
[0 1 0]
[0 0 1]
```

```
L
```

```
[ 1  0  0]
[1/2 1  0]
[ 0  0  1]
```

```
U
```

```
[ 2  1  0]
[ 0 1/2  0]
[ 0  0  1]
```

```
D = M([2,0,0,0,1/2,0,0,0,1])
Uprime = D.inverse()*U
Uprime
```

```
[ 1 1/2 0]
[ 0 1 0]
[ 0 0 1]
```

```
B == L*D*Uprime
```

```
True
```

```
L.transpose() # this is the neat part
```

```
[ 1 1/2 0]
[ 0 1 0]
[ 0 0 1]
```

### Permutations and Pivots

We have seen that elimination sometimes requires us to perform a row operation of swapping the position of two rows to put a pivot in a good place. At first, we want to do this to avoid a zero. But for computational reasons, a machine really likes to have a *big* number as a pivot. So software often uses rows swaps even when not strictly needed.

If all we care about is finding the reduced row echelon form (rref), then this won't worry us. You do whatever operations you want, and the rref is always the same thing. But if we want to keep track with matrices, things get a little complicated.

Here is the important stuff to remember:

1. A row swap is performed by a permutation matrix. A permutation matrix is a matrix with exactly one 1 in each column and in each row. These matrices have the important property that their transposes and their inverses are equal. That is, if  $P$  is a permutation matrix, then  $P^T$  is equal to  $P^{-1}$ . (Not every matrix with this extra property is a permutation matrix. Be careful.)
2. It is possible to figure out what all of the row swaps should be, and then rearrange all of the matrices in an LU decomposition routine. If you do it correctly, you get:

$$P'A = LU$$

or

$$A = PLU$$

where  $P'$  and  $P$  are permutation matrices.

Note: Strang prefers to write things as  $P'A = LU$ , but Sage writes  $A = PLU$ . Fortunately, there is a simple relationship here. Strang's  $P'$  is the transpose (and hence the inverse!) of Sage's  $P$ .

If you haven't figured it out by now, I think that row reduction by hand is really for chumps. Sage (or whatever computational tool you use) makes it waaaaaaaay easier.

```
# using 'partial pivoting' where we get "big pivots"
P, L, U = A.LU()
x = '{0!r}\n\n{1!r}\n\n{2!r}'.format(P,L,U)
print x # fancy python tricks for readable display
```

```
[0 1 0]
[0 0 1]
[1 0 0]

[ 1 0 0]
```

$$\begin{bmatrix} 1/7 & 1 & 0 \\ 4/7 & 1/2 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 7 & 8 & 9 \\ 0 & 6/7 & 12/7 \\ 0 & 0 & 0 \end{bmatrix}$$

```
P*L*U
```

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

```
A == P*L*U
```

```
True
```

```
P.transpose()*A == L*U
```

```
True
```

```
P.transpose()*A
```

$$\begin{bmatrix} 7 & 8 & 9 \\ 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix}$$

## Exercises

Keep this in mind. The computations are simple, but tedious. Perhaps you want to use an appropriate tool.

**Task 2.46.** Find an example of a matrix  $A$  such that  $A^T A = 0$ , but  $A \neq 0$ .

**Task 2.47.** These are true or false questions. If the statement is true, explain why you know it is true. If the statement is false, give an example that shows it is false.

1. The block matrix  $\begin{pmatrix} A & 0 \\ 0 & A \end{pmatrix}$  is automatically symmetric.
2. If  $A$  and  $B$  are symmetric, then their product  $AB$  is symmetric.
3. If  $A$  is not symmetric, then  $A^{-1}$  is not symmetric.

**Task 2.48.** If  $P_1$  and  $P_2$  are permutation matrices, then so is  $P_1 P_2$ . Give examples with  $P_1 P_2 \neq P_2 P_1$  and  $P_3 P_4 = P_4 P_3$ .

**Task 2.49.** Explain the following phenomena in terms of row operations.

1. For any permutation matrix  $P$ , it is the case that  $P^T P = I$ .
2. All row exchange matrices are symmetric:  $P^T = P$ . (other permutation matrices may or may not be symmetric.)
3. If  $P$  is a row exchange matrix, then  $P^2 = I$ .

**Task 2.50.** For each of the following, find an example of a  $2 \times 2$  symmetric matrix with the given property:

1.  $A$  is not invertible.

2.  $A$  is invertible but cannot be factored into  $LU$ .
3.  $A$  can be factored into  $LDL^T$ , but not into  $LL^T$  because  $D$  has negative entries.

**Task 2.51.** This is a new factorization of  $A$  into triangular times symmetric:

Start with  $A = LDU$ . Then  $A = BS$ , where  $B = L(U^T)^{-1}$  and  $S = U^T D U$ .

Explain why this choice of  $B$  is lower triangular with 1's on the diagonal. Explain why  $S$  is symmetric.



## 2.9 Matrix Algebra: Going Further

### The Assignment

- Go back through the exercises in the last four sections of this chapter. Complete any items you did not complete the first time through. Prepare any that we have not discussed in class so that you will be ready to present them.

### Discussion

Now we take a short break to revisit and consolidate the learning you have done so far. Revisit the reading and the exercises you have done in Chapter Two: Linear Equations. The important feature of this work should be learning to think about your own thinking. This sort of **meta-cognition** characterizes expert learners. Eventually, you want to be able to monitor your work at all times and recognize when you understand deeply and when you do not. This will allow you to self-correct.

To help you get started with meta-cognition, I listed learning goals in each section. To go further, you need to explicitly go through the process of reviewing what you can do and what you cannot. Here are some prompts to help you get started with this process.

- Review the learning goals from each section. Can you do the things described? Can you do them sometimes, or have you mastered them so you can do them consistently?
- Look through all of the tasks and go deeper into them. Can you connect each exercise to one of our pictures? Try to build a mental model of how the exercise and its solution work.
- If your first solution to an exercise involve a “guess-and-check” approach, can you now complete the exercise in a *purposeful* and systematic manner?
- Make a list of concepts or exercises that are not clear to you. Phrase each item in your list as a question, and make each question as specific as possible. Talk with fellow students or your instructor until you can answer your own questions.



## Chapter 3

# Vector Spaces and Subspaces

In the last chapter we took on the central problem of linear algebra: solving systems of linear equations. We built a lot of useful tools, but we only answered the questions in the case of *square systems* where the number of variables is equal to the number of equations.

Now it is time to solve the problem in general. As a reminder, the set of questions is this:

- What is the set of solutions to a given system of linear equations?
- When does a given system have solution, and when does it not?
- If there is a solution, how many solutions are there?
- What ways do we have of describing the collection of solutions?
- Is there a computationally effective way to find those solutions?

We will run into the need for a few new tools, each of which has been hiding in the background in our work so far. For square systems, we could get away without this level of detail, but now we will need these concepts to clarify our work: **subspaces**, **rank** and **reduced row echelon form**, and **bases**.

This chapter concludes with a description of the *four fundamental subspaces* associated to a matrix. These will help us put together an understanding of the transformational model of an  $m \times n$  matrix as a function with domain  $\mathbb{R}^n$  and target  $\mathbb{R}^m$ .

### 3.1 Subspaces of $\mathbb{R}^n$

#### The Assignment

- Read Chapter 3 section 1 of *Strang*.
- Read the following and complete the exercises below.

#### Learning Goals

Before class, a student should be able to:

- Identify the column space of a matrix.
- Decide if a set in  $\mathbb{R}^n$  is a subspace or not.

Some time after class, a student should be able to:

- Use the notion of column space to decide if a given linear system has a solution or not.

### Discussion: Vector Spaces and Subspaces

The key concepts in this section are those of a **vector space** and of a **subspace**. The basic idea is that a vector space is a kind of place where the basic operations involved in a linear combination make sense. There is a set of rules for being a vector space, but they are all aimed at the fact that there are two operations (addition and scalar multiplication) and we can form linear combinations with them that make sense.

The biggest thing is that we possibly enlarge the kinds of things we call “vectors.” To a professional mathematician, a vector is anything that is an element of some vector space under consideration. My favorites are things like this:

- The set  $M_{2,2}$  of  $2 \times 2$  matrices is a vector space. But now the things we call vectors are actually matrices.
- The set  $\mathcal{C}(\mathbb{R})$  of continuous functions with domain and range both equal to the set of real numbers is a vector space. But now the things we call vectors are actually functions.
- The set  $\ell(\mathbb{R})$  of sequences  $(x_1, x_2, x_3, x_4, \dots)$  of real numbers is a vector space. But now the things we call vectors are actually whole infinite sequences.

In more advanced mathematics, vector spaces are extremely important. They show up everywhere. But for us, we will mostly stick to the family of vector spaces  $\mathbb{R}^n$ . It will be much more important for us to understand subspaces.

The idea of a subspace is some subset, some part, of a vector space which is a vector space in its own right. The prototype is the  $xy$ -plane inside of  $\mathbb{R}^3$ .

For now, the most important subspaces we see will be derived from individual matrices. Our first example is the column space of a matrix  $A$ . If  $A$  is an  $m \times n$  matrix, then the column space  $\text{col}(A)$  of  $A$  is the collection of  $n$ -vectors which can be expressed as linear combinations of the columns of  $A$ . This is our first exposure to the idea of a **span**. The column space of  $A$  is the subspace of  $\mathbb{R}$  spanned by the columns of  $A$ .

### Sage and subspaces

Sage allows you to construct different kinds of vector spaces. The most important is the standard vector space  $\mathbb{R}^n$ .

```
V = VectorSpace(QQ, 3)
V
```

Vector space of dimension 3 over Rational Field

This is like the `MatrixSpace` construction we have seen before. `V` is now the collection of all vectors of size 3 with rational numbers as entries.

More important for us is the fact that Sage knows how to find the subspaces associated to matrices.

```
A = matrix(QQ, 2, 3, [4, 5, 3, 0, 1, 1])
A
```

```
[4 5 3]
[0 1 1]
```

```
A.column_space()
```

```
Vector space of degree 2 and dimension 2 over Rational Field
Basis matrix:
[1 0]
[0 1]
```

Note that Sage gives information in terms of a basis. We will talk about this concept soon.

```
B = matrix(QQ, 3, 3, [1,2,3,4,5,6,7,8,9])
B
```

```
[1 2 3]
[4 5 6]
[7 8 9]
```

```
B.column_space()
```

```
Vector space of degree 3 and dimension 2 over Rational Field
Basis matrix:
[ 1  0 -1]
[ 0  1  2]
```

This last bit here is interesting. Our original matrix has size  $2 \times 3$ , so the columns are 3-vectors. But this basis is displayed as a matrix where the columns are size 2. What is going on?

Sage prefers to display vectors as rows! This is a big internal preference. We just have to get over it. Well, we have to *remember it* and get over it. To access the basis vectors and display them as columns, we will use the following:

```
B.column_space().basis()[0].column() # first basis vector
```

```
[ 1]
[ 0]
[-1]
```

```
B.column_space().basis()[1].column() # second basis vector
```

```
[0]
[1]
[2]
```

## Exercises

**Task 3.1.** Find an example of a vector which is not in the column space of the matrix

$$A = \begin{pmatrix} 2 & 1 \\ 1 & 1 \end{pmatrix},$$

or explain why that is not possible.

**Task 3.2.** Find an example of a vector which is not in the column space of the matrix

$$B = \begin{pmatrix} 3 & 2 \\ 6 & 4 \end{pmatrix},$$

or explain why it is not possible.

**Task 3.3.** Let  $\mathcal{P}$  be the set of polynomials of degree 3 or less. Explain why  $\mathcal{P}$  is a vector space, or explain why it is not.

**Task 3.4.** Consider the vector space  $\mathbb{R}^2$ . Explain why the following are not subspaces:

- The unit circle.
- The line  $x + y = 4$ .
- The union of lines  $2x + 3y = 0$  and  $x - y = 0$ .
- The first quadrant where  $x \geq 0$  and  $y \geq 0$ .

**Task 3.5.** Let  $W$  be the set of functions  $f$  in  $\mathcal{C}(\mathbb{R})$  which satisfy the differential equation  $f''(x) = 0$ . Decide if  $W$  is a subspace of  $\mathcal{C}(\mathbb{R})$ , and explain your thinking.

**Task 3.6.** Design a  $3 \times 2$  matrix whose column space does not contain the vector  $\begin{pmatrix} 4 \\ 4 \end{pmatrix}$ , or explain why this is not possible.

## 3.2 The Nullspace

### The Assignment

- Read chapter 3 section 2 of *Strang*.
- Read the following and complete the exercises below.

### Learning Goals

Before coming to class, a student should be able to:

- Find the special solutions associated to a matrix.
- Compute the **reduced row echelon form** of a matrix.
- Identify the free variables and the pivot variables associated to a matrix.
- Describe the nullspace of a matrix with the “simplest” set of equations possible.

Some time after class, a student should be able to:

- Describe the complete solution to a homogeneous system of equations  $Ax = 0$ .
- Give an argument for why the nullspace is a vector subspace.

### Discussion: The Nullspace, RREF, and Special Solutions

Strang aims right at the heart of things in this section, and does not waste any space. Let me highlight a few things:

Let  $A$  be an  $m \times n$  matrix. We do not require that  $A$  is a square matrix, that is, it may not be the case that  $m = n$ . What we are most interested in is solving the equation  $Ax = 0$ . Note that if  $A$  is not square, then the vectors  $x$  and  $0$  have different sizes.

By the way, an equation like  $Ax = 0$  where the right hand side is zero is called a **homogeneous equation**.

- The nullspace of  $A$  is the set of vectors  $x$  such that  $Ax = 0$ . It is a theorem that this is a vector subspace of  $\mathbb{R}^n$ . It is common to use the synonym **the kernel of  $A$**  in place of the terminology **the nullspace of  $A$** .
- The key to everything is to not mind that your matrix isn’t square. Just do Gauss-Jordan elimination anyway. The end result is called the **reduced row echelon form** of the matrix, or RREF for short.
- Note that there is no discussion of using an augmented matrix in this section, even though we are solving systems of equations  $Ax = 0$ . This is because the vector on the right hand side is zero and will stay zero. There is nothing interesting to track!
- The RREF is good for several things. One that is often overlooked is that one can use it to rewrite the system of equations in a “simplest form.” I mean that the equations left over in the RREF are somehow the easiest equations to use to cut out the solution as an intersection of hyperplanes.
- In the RREF, all of the structure can be inferred from the splitting of the columns into two types: the pivot columns and the free columns. Since each column is associated to a variable in our system of linear equations (the columns hold coefficients!), it is also common to refer to pivot variables and free variables.

- The number of free columns basically determines the “size” of the nullspace. This is an entry point to the concept of the **dimension** of a vector space. We shall see this in more detail later.
- Strang points out an easy way to find some individual vectors in the null space: he calls these the **special solutions**. This is because they are solutions to the equation  $Ax = 0$ .

## Sage and the Nullspace

The main idea in this section is to get solutions to homogeneous systems  $Ax = 0$  by using the reduced row echelon form. This will compute the nullspace of the matrix  $A$ . Sage has some useful built in commands for this. Let us explore an example.

```
entries = [1,2,3,4,5,6,7,8,9,10,11,12]
A = matrix(QQ, 3,4, entries); A
```

```
[ 1  2  3  4]
[ 5  6  7  8]
[ 9 10 11 12]
```

The most direct way is to ask Sage for the nullspace. But Sage doesn’t call it that. It uses the synonym **kernel**. Also, because we do matrix vector multiplication with vectors on the right, we have to tell Sage to do it that way.

```
A.right_kernel()
```

```
Vector space of degree 4 and dimension 2 over Rational Field
Basis matrix:
[ 1  0 -3  2]
[ 0  1 -2  1]
```

Notice that Sage returns the nullspace by specifying a basis. This is a complete set of special solutions! How can we be sure?

Let’s check that those vectors are actually in the nullspace. This command should return two zero vectors in  $\mathbb{R}^3$ .

```
sage1 = vector([1,0,-3,2])
sage2 = vector([0,1,-2,1])
A*sage1, A*sage2
```

```
((0, 0, 0), (0, 0, 0))
```

Remember that the method to find the reduced row echelon form is `.rref()`.

```
A.rref()
```

```
[ 1  0 -1 -2]
[ 0  1  2  3]
[ 0  0  0  0]
```

What this means is that the null space is described by the two equations

$$\begin{cases} v - y - 2z = 0 \\ x + 2y + 3z = 0 \end{cases}$$

That is, the nullspace is the intersection of these two hyperplanes in  $\mathbb{R}^4$ .

```
# Let's see what Sage's solve command tells us:
v,x,y,z = var('v_x_y_z')
expr1 = v - y - 2*z == 0
expr2 = x + 2*y + 3*z == 0
solve([expr1, expr2], [v,x,y,z])
```



```
[[v == 2*r5 + r6, x == -3*r5 - 2*r6, y == r6, z == r5]]
```

Aha! Sage is trying to tell us that the pivot variables  $v$  and  $x$  should be written in terms of the free variables  $y$  and  $z$ . Strang's "easy way out" is to choose the free variables to be 0's and 1's in combination. Here we have two of them, so we will alternate. One special solution will be to choose  $y = 1$  and  $z = 0$ . Then we solve to get  $v = 1$  and  $x = -2$ .

```
special1 = vector([1,-2,1,0])
# let's check
A*special1
```

```
(0, 0, 0)
(0, 0, 0)
```

Similarly, we can choose  $y = 0$  and  $z = 1$  to get a second special solution. This leads to  $v = 2$  and  $x = -3$ .

```
special2 = vector([2,-3,0,1])
#let's check
A*special2
```

```
(0, 0, 0)
```

These two vectors form a **basis** for the nullspace. Every vector in the nullspace can be written as a linear combination of these two vectors.

What is weird is that Sage always wants its basis to look like 1's and 0's at the beginning, and our process makes them look like that at the end!

How can we be sure everything lines up? Well, it is possible to express the two vectors that Sage gives us as a linear combination of the ones we just found, and vice versa. So we are getting *two* descriptions of the same space.

```
sage1 == -3* special1 + 2* special2
```

```
True
```

```
sage2 == -2* special1 + special2
```

```
True
```

```
special1 == sage1 - 2* sage2
```

```
True
```

```
special2 == 2* sage1 - 3* sage2
```

```
True
```

Can you see the matrix and its inverse hiding behind that? I found those relationships using row operations and a matrix inverse.

Anyway, what has happened is that Sage has performed our calculations above, and then taken the extra steps of putting the matrix of basis vectors into reduced row echelon form, too.

## Exercises

For the first three exercises, your job is to find both

- The minimal set of equations which cuts out the nullspace of the given matrix as an intersection of hyperplanes; and
- the size of the nullspace, expressed through the number “independent directions” it contains.

As always, it will be crucial to explain how you know you are correct.

**Task 3.7.** Consider the matrix

$$A = \begin{pmatrix} 1 & 2 & 3 \\ 4 & 8 & 5 \\ -1 & -2 & 0 \end{pmatrix}$$

**Task 3.8.** Consider the matrix

$$A = \begin{pmatrix} 2 & 1 & 3 & 7 \\ 1 & 1 & 1 & -3 \end{pmatrix}$$

**Task 3.9.** Consider the matrix

$$A = \begin{pmatrix} 2 & 3 & 5 & 6 \\ 1 & 1 & 2 & 3 \\ 0 & 1 & 1 & 0 \\ -1 & -1 & -2 & -3 \\ 1 & 1 & 2 & 3 \end{pmatrix}$$

For the next three exercises, your job is to find a complete set of special solutions to the homogeneous equation  $Ax = 0$ . By a “complete set,” we mean “enough special solutions so that any vector in the nullspace of  $A$  can be expressed as a linear combination of your vectors.”

As always, it will be crucial to explain how you know you are correct.

**Task 3.10.** Consider the matrix

$$A = \begin{pmatrix} 6 & 7 \\ 7 & 8 \\ 1 & 0 \\ 4 & 5 \end{pmatrix}$$

**Task 3.11.** Consider the matrix

$$A = \begin{pmatrix} 23 & 17 & 9 & 2 \\ 1 & -2 & 4 & 1 \\ 22 & 19 & 5 & 1 \end{pmatrix}$$

**Task 3.12.** Consider the matrix

$$A = \begin{pmatrix} -3 & -6 & -9 & -12 \\ 1 & 2 & 3 & 4 \\ 1 & 4 & 9 & 16 \\ 1 & 8 & 27 & 64 \\ -1 & -1 & -1 & -1 \\ 0 & 0 & 1 & 0 \end{pmatrix}$$

### 3.3 Rank and the RREF

### 3.4 Solving a System

## 3.5 Going Further with Solving Systems

### 3.6 Bases

## 3.7 The Four Subspaces

### 3.8 Going Further with the Fundamental Theorem



## Chapter 4

# Orthogonality

Here is some introductory text.



## Chapter 5

# Determinants

Here is some introductory text.



## Chapter 6

# Eigendata and the Singular Value decomposition

Here is some introductory text.

